Chapter 10
Nonlinear Structural Analysis

The nonlinear load-displacement relationship—the stress-strain relationship with a nonlinear function of stress, strain, and/or time; changes in geometry due to large displacements; irreversible structural behavior upon removal of the external loads; change in boundary conditions such as a change in the contact area and the influence of loading sequence on the behavior of the structure—requires a nonlinear structural analysis. The structural nonlinearities can be classified as geometric nonlinearity, material nonlinearity, and contact or boundary nonlinearity.

The governing equations concerning large deformations are nonlinear with respect to displacements and velocities. The material behavior can be linear or nonlinear, and the boundary conditions can also exhibit nonlinearity. Geometrical nonlinearity arising from large deformations is associated with the necessity to distinguish between the coordinates of the initial and final states of deformation, and also with the necessity to use the complete expressions for the strain components. The material can exhibit either time-dependent or time-independent nonlinear behavior. Nonlinearity due to boundary conditions emerges from a nonlinear relationship between the external forces and the boundary displacements. The presence of contact conditions also leads to a nonlinear structural analysis because the extent of the contact region and the contact stresses are not known a priori.

The solution to the nonlinear governing equations can be achieved through an incremental approach. The incremental form of the governing equations can be written as

$$\mathbf{K}(\mathbf{u})\Delta\mathbf{u} = \Delta\mathbf{P} \tag{10.1}$$

in which $\Delta\mathbf{u}$ and $\Delta\mathbf{P}$ represent the unknown incremental displacement vector and the known incremental applied load vector, respectively. The solution is constructed by taking a series of linear steps in the appropriate direction in order to closely approximate the exact solution. Depending on the nature of the nonlinearity, the magnitude of each step and its direction may involve several iterations. The computational algorithms and the associated parameters must be chosen with extreme care. The solution to nonlinear problems may not be unique.

The online version of this book (doi: 10.1007/978-1-4939-1007-6_10) contains supplementary material, which is available to authorized users

© Springer International Publishing 2015

E. Madenci, I. Guven, *The Finite Element Method and Applications in Engineering Using ANSYS®*, DOI 10.1007/978-1-4899-7550-8_10

When solving nonlinear problems, ANSYS uses the Newton-Raphson (N-R) method, which involves an iterative procedure. This method starts with a trial (assumed) solution, $\mathbf{u} = \mathbf{u}_i$, to determine the magnitude of the next step (increment), $\Delta\mathbf{u}_i = \mathbf{K}^{-1}(\mathbf{u}_i)\Delta\mathbf{P}$, and the corresponding *out-of-balance load vector*, $\Delta\mathbf{R}_i = \Delta\mathbf{P} - \mathbf{K}(\mathbf{u}_i)\Delta\mathbf{u}_i$, which is the difference between the applied loads and the loads evaluated based on the assumed solution. In order to satisfy the equilibrium conditions exactly, *the out-of-balance load vector* must be zero. However, as the nonlinear equilibrium conditions are solved approximately, a tolerance is introduced for the out-of-balance load vector in order to terminate the solution procedure. In each iteration, the N-R method computes the out-of-balance load vector and checks for convergence based on the specified tolerance. If the convergence criterion is not satisfied, the trial solution is updated as $\mathbf{u}_{i+1} = \mathbf{u}_i + \Delta\mathbf{u}_i$ based on the calculated incremental displacements, and the next incremental solution vector is determined as $\Delta\mathbf{u}_{i+1} = \mathbf{K}^{-1}(\mathbf{u}_{i+1})\Delta\mathbf{P}$ leading to the computation of the new out-of-balance load vector $\Delta\mathbf{R}_{i+1} = \Delta\mathbf{P} - \mathbf{K}(\mathbf{u}_{i+1})\Delta\mathbf{u}_{i+1}$; this procedure is repeated until convergence is accomplished.

Several methods for improving the convergence (or convergence rate) are available in ANSYS. These include automatic time stepping, a bisection method, and line search algorithms. The user may choose to have full control or let ANSYS choose the options.

In a nonlinear solution in ANSYS, there are three distinct levels: (1) *Load Steps*, (2) *Substeps*, and (3) *Equilibrium Iterations*. The number of load steps is specified by the user. Different load steps must be used if the loading on the structure changes abruptly. The use of load steps also becomes necessary if the response of the structure at specific points in time is desired. A solution within each load step is obtained by applying the load incrementally in substeps. Within each substep, several equilibrium iterations are performed until convergence is accomplished, after which ANSYS proceeds to the next substep. As the number of substeps used increases, the accuracy of the solution improves. However, this also means that more computational time is being used. ANSYS offers the *Automatic Time Stepping* feature to optimize the task of obtaining a solution with acceptable accuracy in a reasonable amount of time. The automatic time stepping feature decides on the number and size of substeps within load steps. When using automatic time stepping, if a solution fails to converge within a substep, the *bisection* method is activated, which restarts the solution from the last converged substep.

The ANSYS program has default values for all of the nonlinear *solution controls*, including the convergence options. The **`SOLCONTROL`** command is used to turn these defaults on or off. The help page for the **`SOLCONTROL`** command provides a comprehensive list of the default values of nonlinear analysis settings when solution controls are on (**`SOLCONTROL, ON`**), which is the default setting. It is also possible to modify specific controls while leaving the rest for ANSYS to assign. Some of the commonly used commands for modifying/specifying nonlinear analysis settings with brief descriptions are:

`AUTOTS` Command: Turns automatic time stepping on or off.

DELTIM Command: Specifies time step size and/or minimum and maximum time step sizes to be used within a load step.

NSUBST Command: Specifies number of substeps and/or minimum and maximum number of substeps to be used within a load step.

NEQIT Command: Specifies maximum number of equilibrium iterations within a substep. If this number is reached with no converged solution, and if automatic time stepping is on, then ANSYS employs the *bisection* method to achieve convergence. Otherwise, the solution is terminated.

KBC Command: Specifies whether the loads are interpolated (ramped) for each substep from the values of the previous load step to the values of the current load step.

EQSLV Command: Specifies the type of solver to be used to solve the matrix system of equations. By default, it is the *Sparse Solver*; however, there are several other solvers available that may be more efficient for the particular problem being solved.

CNVTOL Command: Specifies convergence tolerance values for the nonlinear analyses.

NROPT Command: Specifies which type of Newton-Raphson method is used in the solution.

LNSRCH Command: Specifies whether a line search is to be used with the Newton-Raphson method in the solution.

PRED Command: Specifies whether a predictor algorithm is to be used in the solution.

ARCLEN Command: Toggles the *arc-length* method on or off.

SSTIF Command: The stiffness of certain materials increases with the increased stress levels within the structure (e.g., cables and membranes). This command toggles the *stress stiffening* effects on or off.

TIMINT Command: Toggles the transient effects on or off.

OUTRES Command: Specifies the amount and frequency of the data saved in the results file. By default, results associated with the last substep of each load step are written in the results file.

The commands described above require special attention, which may be crucial to the success of the analysis. There are no “golden standard” values for time step sizes, the number of equilibrium iterations, or the number of substeps. The user accumulates knowledge on the use of these features with every new analysis. It is highly recommended that the user consult the ANSYS Help pages on nonlinear analysis and individual commands, which provide detailed guidelines. Some general suggestions on achieving success in nonlinear analyses are as follows:

Nonlinear analyses require more computational time. Therefore, when solving nonlinear static problems, it may be helpful to solve a preliminary version of the problem with no nonlinearities. The results from the *linear solution* may indicate mistakes in modeling, meshing, and application of boundary conditions in a shorter time frame. Also, the linear solution provides information about the regions where high stress gradients are expected, thus guiding the user to modify the mesh (make it more refined) in those regions.

In nonlinear analyses, it is important to utilize all possible simplifications in order to improve convergence and reduce the computational cost. For example, if the problem can be simplified as a plane stress or plane strain idealization, then the user should take advantage of this opportunity.

Reading the contents of the *Output Window* and the *Error File* (`jobname.err`) is crucial in finding the specific reason why the solution does not converge.

Another important consideration in dealing with nonlinear problems is the *path dependency* of the solution. When all the materials in a problem exhibit linear behavior, the order in which the loads are applied does not make any difference in the results. However, when a nonlinear material behavior is present, results obtained by applying the same set of loads in different orders may differ from each other.

Detailed step-by-step instructions for numerous example problems are provided in Chaps. 6, 8, and 9. The command line equivalents of each of these example problems, written in the ANSYS Parametric Design Language (APDL), are included in the accompanying CD-ROM. The use of the APDL is described in Chap. 7.

In this chapter, the nonlinear structural analyses arising from (i) geometric nonlinearity, (ii) material nonlinearity, and (iii) contact conditions are considered in order to demonstrate the nonlinear features of ANSYS. However, APDL is chosen to be the main method of interacting with ANSYS because of its versatility and efficiency. Explanations are included in each command line after an exclamation mark (`!`). Step-by-step instructions are sporadically given when they are considered to be beneficial to the user. It is highly recommended that the user have a good understanding of APDL before delving into this chapter.

10.1 Geometric Nonlinearity

Geometric nonlinearities arise from the presence of large strain, small strains but finite displacements and/or rotations, and loss of structural stability. Large strains, over 5 % may occur in rubber structures and metal forming. Slender structures such as bars and thin plates may experience large displacements and rotations with small strains. Initially stressed structures with small strains and displacements may undergo a loss of stability by buckling.

Two problems are considered. The first problem involves a thin cantilever plate subjected to a point load at one of the free corners. Because the plate is thin, the resulting displacement components are in comparable order to its geometric dimensions, thus the geometry changes. This requires the stiffness matrix to be modified by accounting for the changes in the geometry. Results are compared to the solution of the same problem obtained by disregarding the nonlinear geometry effects.

The second problem involves a composite plate with a circular hole subjected to compression. As the applied loading increases, the plate is expected to buckle. Eigenvalue Buckling Analysis is one of the methods that could be used to solve this problem (as explained in Chap. 8). However, Eigenvalue Buckling Analysis evaluates only the buckling load; it does not solve for the events after buckling occurs. Alternatively, nonlinear geometry effects are turned on and post-buckling behavior is evaluated along with the buckling load.



Fig. 10.1 Cantilever plate with a transverse force at one corner

10.1.1 Large Deformation Analysis of a Plate

Consider the cantilever plate with a transverse force at one corner shown in Fig. 10.1. The plate has a length, width, and thickness of 40, 30, and 0.4 m, respectively. Its elastic modulus is 120 MPa, and the Poisson's ratio is 0.3. The maximum applied load of 40 kN is reached in five equal increments. The nonlinear geometry option is used in the ANSYS solution. This is achieved by writing *Load Step Files* for each increment, and obtaining the solution from these files (**LSSOLVE** command). The goal is to find the displacement components as the applied load increases, at points A and B shown in Fig. 10.1.

Model Generation

Element type **SHELL181** is used in the analysis. The number of element divisions on all of the lines is specified as 20, and mapped meshing is utilized. The following command input is used for model generation:

```
/PREP7                  ! ENTER PREPROCESSOR
ET,1,181                ! SPECIFY ELEMENT TYPE AS SHELL 181
MP,EX,1,1.2E8           ! SPECIFY ELASTIC MODULUS
MP,NUXY,1,0.3           ! SPECIFY POISSON'S RATI0
SECT,1,SHELL            ! SPECIFY PLATE THICKNESS
SECDATA,0.4,1           ! DEFINE THICKNESS FOR THE PLATE
K,1,0                   ! CREATE KEYPOINTS
K,2,40                  !
K,3,40,30               !
K,4,0,30                !
A,1,2,3,4               ! CREATE AREA THROUGH KEYPOINTS
LESIZE,ALL,,,20         ! SPECIFY NO. OF DIVISIONS ON LINES
MSHKEY,1                ! ENFORCE MAPPED MESHING
AMESH,ALL               ! MESH THE AREA
FINISH                  ! EXIT PREPROCESSOR
```

Solution

Nonlinear geometry effects are turned on using the **NLGEOM** command. The maximum number of equilibrium iterations is specified as 1000 using the **NEQIT** command. After the specification of displacement constraints, the transverse load is specified in increments of 8000 N. After application of each increment, a load step file is written to the *Working Directory* (with the naming convention: *file.s01*, *file.s02*, **...**) using the **LSWRITE** command. When the **LSWRITE** command is issued, ANSYS writes all of the specified boundary conditions to the load step file. When finished with the last load step (40,000 N), the **LSSOLVE** command is issued to start the solution by reading the boundary conditions from load step files 1 through 5. The following command input is used for the solution:

```
/SOLU                        ! ENTER SOLUTION PROCESSOR
ANTYPE,STATIC                ! SPECIFY ANALYSIS TO BE STATIC
NLGEOM,ON                    ! TURN NONLINEAR GEOMETRY
                             ! EFFECTS ON
NEQIT,1000                   ! SPECIFY MAX. # OF EQUILIBRIUM
                             ! ITERATIONS
OUTRES,ALL,ALL               ! WRITE ALL SOLUTION ITEMS FOR
                             ! EVERY  SUBSTEP
NSEL,S,LOC,X,0               ! SELECT NODES AT X = 0
D,ALL,ALL                    ! CONSTRAIN ALL DOFS AT SELECTED
                             ! NODES
ALLSEL                       ! SELECT EVERYTHING
                             ! START FIRST LOAD INCREMENT
NSEL,S,LOC,X,40              ! SELECT NODES AT X = 40
NSEL,R,LOC,Y,0               ! SELECT NODES AT Y = 0 FROM THE
                             ! SELECTED SET
F,ALL,FZ,-8000               ! SPECIFY FORCE IN NEG. Z-DIR AT
                             ! SELECTED NODES
ALLSEL                       ! SELECT EVERYTHING
LSWRITE                      ! WRITE LOAD STEP FILE 1
                             ! SECOND LOAD INCREMENT
NSEL,S,LOC,X,40              ! SELECT NODES AT X = 40
NSEL,R,LOC,Y,0               ! SELECT NODES AT Y = 0 FROM THE
                             ! SELECTED SET
F,ALL,FZ,-16000              ! SPECIFY FORCE IN NEG. Z-DIR AT
                             ! SELECTED NODES
ALLSEL                       ! SELECT EVERYTHING
LSWRITE                      ! WRITE LOAD STEP FILE 2
                             ! THIRD LOAD INCREMENT
NSEL,S,LOC,X,40              ! SELECT NODES AT X = 40
NSEL,R,LOC,Y,0               ! SELECT NODES AT Y = 0 FROM THE
                             ! SELECTED SET
F,ALL,FZ,-24000              ! SPECIFY FORCE IN NEG. Z-DIR AT
                             ! SELECTED NODES
ALLSEL                       ! SELECT EVERYTHING
LSWRITE                      ! WRITE LOAD STEP FILE 3
                             ! FOURTH LOAD INCREMENT
NSEL,S,LOC,X,40              ! SELECT NODES AT X = 40
```

```
NSEL,R,LOC,Y,0              ! SELECT NODES AT Y = 0 FROM THE
                            ! SELECTED SET
F,ALL,FZ,-32000             ! SPECIFY FORCE IN NEG. Z-DIR AT
                            ! SELECTED NODES
ALLSEL                      ! SELECT EVERYTHING
LSWRITE                     ! WRITE LOAD STEP FILE 4
                            ! FIFTH LOAD INCREMENT
NSEL,S,LOC,X,40             ! SELECT NODES AT X = 40
NSEL,R,LOC,Y,0              ! SELECT NODES AT Y = 0 FROM THE
                            ! SELECTED SET
F,ALL,FZ,-40000             ! SPECIFY FORCE IN NEG. Z-DIR AT
                            ! SELECTED NODES
ALLSEL                      ! SELECT EVERYTHING
LSWRITE                     ! WRITE LOAD STEP FILE 5
LSSOLVE,1,5,1               ! SOLVE FROM LOAD STEP FILES
FINISH                      ! EXIT SOLUTION PROCESSOR
```

Postprocessing

In the postprocessing, node numbers for the nodes at points A and B are stored in parameters **NA** and **NB**, respectively. A do loop is used for extraction of the results data at different results sets corresponding to the load steps. The commands **/OUTPUT** and ***VWRITE** are used for redirecting output and writing parameters to external files. The use of shortcuts for ***GET** functions **UX(*N*)**, **UY(*N*)**, and **UZ(*N*)**, is also demonstrated. These functions retrieve the *x*-, *y*-, and *z*-displacements of node ***N***. The following command input is used for postprocessing:

```
/POST1                           ! ENTER POSTPROCESSOR
NSEL,S,LOC,X,40                  ! SELECT NODES AT X = 40
NSEL,R,LOC,Y,0                   ! RESELECT NODE AT Y = 0
*GET,NA,NODE,0,NUM,MIN           ! STORE NODE # OF PT A
                                 ! INTO NA
NSEL,S,LOC,X,40                  ! SELECT NODES AT X = 40
NSEL,R,LOC,Y,30                  ! RESELECT NODE AT Y = 30
*GET,NB,NODE,0,NUM,MIN           ! STORE NODE # OF PT B
                                 ! INTO NB
ALLSEL                           ! SELECT EVERYTHING
*DO,I,1,5                        ! START DO LOOP
SET,I                            ! SET RESULTS TO LOAD STEP I
/OUTPUT,ADISP,OUT,,APPEND        ! REDIRECT OUTPUT TO FILE
                                 ! ADISP.OUT
*VWRITE,UX(NA),UY(NA),UZ(NA)     ! WRITE DISPLACEMENTS TO FILE
(E16.8,5X,E16.8,5X,E16.8,5X)     ! FORMAT STATEMENT
/OUTPUT                          ! REDIRECT OUTPUT TO OUTPUT
                                 ! WINDOW
/OUTPUT,BDISP,OUT,,APPEND        ! REDIRECT OUTPUT TO FILE
                                 ! BDISP.OUT
*VWRITE,UX(NB),UY(NB),UZ(NB)     ! WRITE DISPLACEMENTS TO FILE
(E16.8,5X,E16.8,5X,E16.8,5X)     ! FORMAT STATEMENT
/OUTPUT                          ! REDIRECT OUTPUT TO
                                 ! OUTPUT WINDOW
*ENDDO                           ! END DO LOOP
```



Fig. 10.2 Variation of displacement components at points *A* and *B* as the load increases incrementally

After execution of the command input given above, the *x*-, *y*-, and *z*-displacements at points A and B are written to files *ADISP.OUT* and *BDISP.OUT*, respectively. Figure 10.2 shows the variation of displacement components as the load increases incrementally. Also plotted in Fig. 10.2 are the *z*-displacements of points A and B obtained by disregarding geometric nonlinearity (indicated in the legend as "linear"). Displacements in the *x*- and *y*-directions are identically zero for this case.

10.1.2 Post-buckling Analysis of a Plate with a Hole

Consider a 9.5-in-square composite plate with a circular hole of radius 1.5 in, as illustrated in Fig. 10.3. The laminate lay-up is $[\pm 30]_{3s}$, with a total of 12 layers, symmetric with respect to the mid-plane, and the orientation of the layers alternates between 30 and −30. Each layer has moduli of $E_L = 18.5 \times 10^6$ psi, $E_T = 1.6 \times 10^6$ psi, and $G_{LT} = 0.832 \times 10^6$ psi, and a Poisson's ratio of $\nu_{LT} = 0.35$. Each layer is 0.01 in. thick, resulting in a total laminate thickness of 0.12 in.

Along the right edge of the laminate, an axial concentrated load of 12,000 lb is introduced through a rigid end. This type of load introduction, requiring the *x*-displacement to be uniform, is enforced by coupled degrees of freedom, as explained in Chap. 11. All degrees of freedom (displacements and rotations) are constrained along the left edge. Along the horizontal edges, in-plane displacements and rotations about these edges are permitted. In order to trigger the nonlinear response, a



Fig. 10.3 Composite plate with a hole

sinusoidal imperfection with an amplitude of 1 % of the total laminate thickness is used as follows:

$$z = 0.012 \sin\left(\frac{\pi x}{9.5}\right)\sin\left(\frac{\pi y}{9.5}\right) \tag{10.2}$$

The goal is to obtain the variations of the z-displacement at points A (4.75, 6.25, 0) and B (6.25, 4.75, 0), and the x-displacement at point C (9.5, 4.75, 0) as the applied load increases (points A, B, and C are indicated in Fig. 10.3).

Model Generation

The element type used in the analysis is **`SHELL181`**, which is a layered element especially useful for modeling composite plates. By default, the element coordinate system is derived from the local geometry of each element. Although this may be convenient in certain cases, in this problem it is required that the element coordinate system for each element be aligned with the global Cartesian system. This is achieved by using **`ESYS`** the command. After specifying real constants (thickness and layer information) and orthotropic material properties, and creating the solid model, the mesh is generated. The mesh has all of its nodes and elements on the x-y plane, i.e., $z = 0$. However, in order to capture the buckling behavior, the flatness of the plate is slightly perturbed by means of a double-sinusoidal surface, as explained earlier. This is achieved by modifying the *z*-coordinates of each node through Eq. (10.2). When a mesh is generated on solid modeling entities (lines, areas and volumes), the resulting nodes and elements are *attached* to solid modeling entities and their modification is not allowed. Therefore, the mesh (nodes and elements) is *detached* from the solid model using the **MODMSH** command, after which the nodal coordinates are modified utilizing a do loop in accordance with Eq. (10.2). The following command input is used for model generation:

```
/PREP7                         ! ENTER PREPROCESSOR
ET,1,SHELL181                  ! SPECIFY ELEMENT TYPE
ESYS,0                         ! ALIGN ELEM CS WITH GLOBAL
                               ! CARTESIAN
MP,EX,1,18.5E6                 ! SPECIFY MATERIAL
                               ! PROPERTIES
MP,EY,1,1.6E6                  !
MP,EZ,1,1.6E6                  !
MP,GXY,1,0.832E6               !
MP,PRXY,1,0.35                 !
MP,GYZ,1,0.533E6               !
MP,PRYZ,1,0.5                  !
MP,GXZ,1,0.832E6               !
MP,PRXZ,1,0.35                 !
SECT,1,SHELL                   ! SPECIFY REAL CONSTANTS
                               ! DEFINE THICKNESS AND
                               ! ORIENTATION FOR LAYER 1-12
SECDATA,0.01,1,30              !
SECDATA,0.01,1,-30             !
SECDATA,0.01,1,30              !
SECDATA,0.01,1,-30             !
SECDATA,0.01,1,30              !
SECDATA,0.01,1,-30             !
SECDATA,0.01,1,30              !
SECDATA,0.01,1,-30             !
SECDATA,0.01,1,30              !
SECDATA,0.01,1,-30             !
SECDATA,0.01,1,30              !
SECDATA,0.01,1,-30             !
K,1,0                          ! CREATE KEYPOINTS
K,2,9.5                        !
K,3,9.5,9.5                    !
K,4,0,9.5                      !
A,1,2,3,4                      ! CREATE SQUARE AREA
CYL4,4.75,4.75,1.5             ! CREATE CIRCULAR AREA
ASBA,1,2                       ! SUBTRACT AREAS
NUMCMP,ALL                     ! COMPRESS ENTITY NUMBERS
LSEL,S,,,1,4                   ! SELECT LINES
```

```
LESIZE,ALL,,,10                    !SPECIFY # OF DIVISIONS
LSEL,S,,,5,8                       !SELECT LINES
LESIZE,ALL,,,5                     !SPECIFY # OF DIVISIONS
AMESH,ALL                          !MESH AREA
MODMSH,DETACH                      !DETACH MESH FROM SOLID
                                   !MODEL
*GET,NNUMBER,NODE,0,NUM,MAX        !GET MAXIMUM NODE # (ALSO #
                                   !OF NODES AS NUMBERS ARE
                                   !COMPRESSED)
PI=4*ATAN(1)                       !DEFINE PI (3.1415...)
*DO,I,1,NNUMBER                    !START DO LOOP ON NODES
*GET,TMPX,NODE,I,LOC,X             !GET X-COORD OF CURRENT
                                   !NODE
*GET,TMPY,NODE,I,LOC,Y             !GET Y-COORD OF CURRENT
                                   !NODE
TMPZ=SIN(PI*NX(I)/9.5)*SIN(PI*NY(I)/9.5)*0.012
                                   !EQ. (10.2)
N,I,NX(I),NY(I),TMPZ               !REDEFINE CURRENT NODE
*ENDDO                             !END DO LOOP
FINISH                             !EXIT FROM PREPROCESSOR
```

Solution

In the solution phase of this problem, first, nonlinear geometry effects are turned on (**NLGEOM** command). Results associated with each substep are written to the results file using the **OUTRES** command. In order to improve accuracy and convergence, the number of substeps is specified as 100 using the **NSUBST** command. Displacement constraints are then applied using the **D** command. The condition that the *x*-displacement be uniform along the right boundary ($x = 9.5$) is imposed by selecting all of the nodes along that boundary and issuing the CP command. This command defines a set of coupled degrees of freedom (DOF) of the selected nodes (in this case the *x*-displacement) and enforces these DOF to be equal. The concept of coupled DOF is described as a separate topic in Sect. 11.1. The following command input is used for the solution:

```
/SOLU                ! ENTER SOLUTION PROCESSOR
ANTYPE,STATIC        ! SPECIFY STATIC SOLUTION
NLGEOM,ON            ! TURN NONLINEAR GEOMETRY ON
OUTRES,ALL,ALL       ! WRITE RESULTS FOR EVERY SUBSTEP
NSUBST,100           ! SET NUMBER OF SUBSTEPS TO BE 100
NSEL,S,LOC,X,0       ! SELECT NODES AT X = 0
D,ALL,ALL            ! CONSTRAIN ALL DOFS
NSEL,S,LOC,Y,0       ! SELECT NODES AT Y = 0
NSEL,A,LOC,Y,9.5     ! ADD TO SELECTION NODES AT Y = 9.5
D,ALL,UZ,0           ! CONSTRAIN Z-DISPLACEMENTS
D,ALL,ROTY,0         ! CONSTRAIN ROTATIONS ABOUT Y-AXIS
NSEL,S,LOC,X,9.5     ! SELECT NODES AT X = 9.5
D,ALL,UZ,0           ! CONSTRAIN Z-DISPLACEMENTS
D,ALL,ROTX,0         ! CONSTRAIN ROTATIONS ABOUT X-AXIS
D,ALL,ROTY,0         ! CONSTRAIN ROTATIONS ABOUT Y-AXIS
```

```
ALLSEL              ! SELECT EVERYTHING
D,ALL,ROTZ,0        ! CONSTRAIN ROTATIONS ABOUT Z-AXIS
NSEL,S,LOC,X,9.5    ! SELECT NODES AT X = 9.5
CP,1,UX,ALL         ! COUPLE X-DISPL. OF SELECTED NODES
NSEL,R,LOC,Y,4.75   ! RESELECT NODE AT Y = 4.75
F,ALL,FX,-12000     ! APPLY TOTAL LOAD IN X-DIRECTION
ALLSEL              ! SELECT EVERYTHING
SOLVE               ! OBTAIN SOLUTION
FINISH              ! EXIT SOLUTION PROCESSOR
```

Postprocessing

Because the loading is applied slow enough that there are no dynamic effects within the structure, referred to as quasi-static, the nonlinear nature of the problem requires that the load is incrementally increased.

Turning on the nonlinear geometry effects using the **NLGEOM** command leads to the problem being solved in small increments (substeps), even though the analysis type is declared as static. ANSYS automatically assigns the value ***1*** as the time at the end of the load step. In nonlinear static solutions, time is a measure of the fraction of the total load applied at the current substep. For example, if time has a value of ***0.2*** at a particular substep, this means that the load applied during solution at that substep is 20 % of the total load. In the following command input, the node numbers of points A, B, and C are stored in parameters **NA**, **NB**, and **NC**, respectively, followed by extraction of the total number of substeps (parameter **SB**). A do loop is set up so that the results associated with every substep can be retrieved and written to external files sequentially. Within the do loop, the time associated with the current substep is extracted (parameter **TT**) for subsequent scaling of the applied load (parameter **TF**), and z-displacements at points A and B and the absolute value of the x-displacement at point C are written to files *AZD.OUT*, *BZD.OUT*, and *CXD.OUT*, respectively. Figure 10.4 shows the variation of these displacements with incremental load. In Fig. 10.4, z-displacements at points A and B are denoted by w(A) and w(B), respectively, and u(C) designates the absolute value of the x-displacement at point C.

```
/POST1                      ! ENTER POSTPROCESSOR
NSEL,S,LOC,X,4.75           ! SELECT NODES AT X = 4.75
NSEL,R,LOC,Y,6.25           ! RESELECT NODE AT Y = 6.25
*GET,NA,NODE,0,NUM,MIN      ! STORE NODE # OF PT A INTO NA
NSEL,S,LOC,X,6.25           ! SELECT NODES AT X = 6.25
NSEL,R,LOC,Y,4.75           ! RESELECT NODE AT Y = 4.75
*GET,NB,NODE,0,NUM,MIN      ! STORE NODE # OF PT B INTO NB
NSEL,S,LOC,X,9.5            ! SELECT NODES AT X = 9.5
NSEL,R,LOC,Y,4.75           ! RESELECT NODE AT Y = 4.75
*GET,NC,NODE,0,NUM,MIN      ! STORE NODE # OF PT C INTO NC
ALLSEL                      ! SELECT EVERYTHING
```



Fig. 10.4 Variation of displacement components at points *A*, *B*, and *C* as the load increases incrementally

```
SET,LAST                       ! SET RESULTS TO LAST SUBSTEP
*GET,SB,ACTIVE,0,SET,SBST      ! STORE # OF SUBSTEPS TO SB
*DO,I,1,SB                     ! START DO LOOP
SET,1,I                        ! SET RESULTS TO SUBSTEP I
*GET,TT,ACTIVE,0,SET,TIME      ! STORE CURRENT TIME TO TT
TF=TT*12000                    ! FIND CURRENT FORCE
/OUTPUT,AZD,OUT,,APPEND        ! REDIRECT OUTPUT TO FILE AZD.OUT
*VWRITE,TF,UZ(NA)              ! WRITE TIME & DISP TO FILE
(E16.8,5X,E16.8)               ! FORMAT STATEMENT
/OUTPUT                        ! REDIRECT OUTPUT TO
                               ! OUTPUT WINDOW
/OUTPUT,BZD,OUT,,APPEND        ! REDIRECT OUTPUT TO FILE BZD.OUT
*VWRITE,TF,UZ(NB)              ! WRITE TIME & DISP TO FILE
(E16.8,5X,E16.8)               ! FORMAT STATEMENT
/OUTPUT                        ! REDIRECT OUTPUT TO
                               ! OUTPUT WINDOW
/OUTPUT,CXD,OUT,,APPEND        ! REDIRECT OUTPUT TO FILE CXD.OUT
*VWRITE,TF,ABS(UX(NC))         ! WRITE TIME & DISP TO FILE
(E16.8,5X,E16.8)               ! FORMAT STATEMENT
/OUTPUT                        ! REDIRECT OUTPUT TO
                               ! OUTPUT WINDOW
*ENDDO                         ! END DO LOOP
```

10.2 Material Nonlinearity

Material nonlinearities arise from the presence of time-independent behavior, such as plasticity, time-dependent behavior such as creep, and viscoelastic/viscoplastic behavior where both plasticity and creep effects occur simultaneously. They may result in load sequence dependence and energy dissipation (irreversible structural behavior).



Fig. 10.5 Aluminum sphere subjected to internal pressure, and corresponding finite element mesh using axisymmetric shell elements (**SHELL208**)

ANSYS offers a wide variety of nonlinear material behavior models, including nonlinear elasticity, hyperelasticity, viscoelasticity, plasticity, viscoplasticity, creep, swelling, and shape memory alloys. Several of these nonlinear material models can be specified in a combined fashion (an exhaustive list of models that can be combined is given in the ANSYS Structural Analysis Guide). In the following subsections, four problems are considered that demonstrate the solution methods involving plasticity with isotropic hardening, viscoelasticity, viscoplasticity with Anand's model, and combined plasticity and creep.

10.2.1 Plastic Deformation of an Aluminum Sphere

Consider a thin-walled aluminum sphere with a radius of $r = 10$ in and a thickness of $t = 0.2$ in, as shown in Fig. 10.5. The sphere is subjected to an internal pressure of $p_0 = 1600\text{psi}$. The elastic modulus and Poisson's ratio of the shell are $E = 10^7$ psi and $\nu = 0.3$, respectively. The plastic behavior of aluminum is governed by

$$\sigma_e = 30000 + 136000(\varepsilon_p)^{1/2} \tag{10.3}$$

in which σ_e is the effective stress and ε_p designates the plastic strain. Figure 10.6 shows the stress vs. total strain ($\varepsilon = \varepsilon_e + \varepsilon_p$, ε_e: elastic strain) curve based on Eq. (10.3). This curve is input in ANSYS by means of a data table for nonlinear material behavior, which is given through data points (see Table 10.1). The goal is to obtain the radial displacements, as well as the strain field.

Model Generation

In order to model the thin-walled aluminum sphere, a 2-noded axisymmetric shell element (**SHELL208**) is used. For the nonlinear material behavior, the multiple-point isotropic hardening rule is chosen (**TB** command with **MISO** option). Twenty data points for strain and stress values are entered using the **TBPT** command [the data point (0, 0) is implied]. The stress value of the first data point defines the yield stress, i.e., $\sigma_{ys} = 30000\text{psi}$. Due to the symmetry conditions, only a quarter circle is modeled.

```
/PREP7                  !ENTER PREPROCESSOR
ET,1,208                !USE SHELL208 ELEMENT
MP,EX,1,1E7             !SPECIFY ELASTIC MODULUS
MP,NUXY,1,0.3           !SPECIFY POISSON'S RATIO
SECT,1,SHELL            !SPECIFY SECTION TYPE
SECDATA,0.2,1           !DEFINE THICKNESS FOR SHELL
TB,MISO,1,1,20,         !SPECIFY MULTILINEAR ISOTROPIC
                        !HARDENING
TBPT,,0.00300,30000     !ENTER STRAIN VS STRESS DATA
                        !POINTS
TBPT,,0.00350,33041     !
TBPT,,0.00400,34300     !
TBPT,,0.00450,35267     !
TBPT,,0.00500,36082     !
TBPT,,0.00550,36800     !
TBPT,,0.00600,37449     !
TBPT,,0.00650,38045     !
TBPT,,0.00700,38601     !
TBPT,,0.00750,39123     !
TBPT,,0.00800,39616     !
TBPT,,0.00850,40086     !
TBPT,,0.00875,40312     !
TBPT,,0.00900,40534     !
TBPT,,0.00925,40751     !
TBPT,,0.00950,40964     !
TBPT,,0.00975,41173     !
TBPT,,0.02400,49708     !
TBPT,,0.04000,56160     !
TBPT,,0.06300,63313     !
K,1,0,0                 !CREATE KEYPOINTS
K,2,10,0                !
K,3,0,10                !
LARC,2,3,1,10           !CREATE ARC
LESIZE,ALL,,,20         !SPECIFY # OF ELEMENTS ON LINE
LMESH,ALL               !MESH LINE
FINISH                  !EXIT PREPROCESSOR
```

Solution

The solution phase of this problem is rather straightforward and involves application of boundary conditions and the internal pressure. No specific solution controls options are specified; ANSYS uses default settings for nonlinear solution.

```
/SOLU                   !ENTER SOLUTION PROCESSOR
P0=1600                 !DEFINE PARAMETER FOR INTERNAL
                        !PRESSURE
NSEL,S,LOC,Y,0          !SELECT NODES AT Y = 0
D,ALL,UY,0              !CONSTRAIN Y-DISPL AT SELECTED NODES
NSEL,S,LOC,Z,0          !SELECT NODES AT Z = 0
SFE,ALL,1,PRES,,P0      !SPECIFY INTERNAL PRESSURE ALONG ELEMS
SOLVE                   !OBTAIN SOLUTION
FINISH                  !EXIT SOLUTION PROCESSOR
```



Fig. 10.6 Nonlinear stress-strain behavior of aluminum

Postprocessing

The elastic and plastic strains along the inner surface of the sphere are stored in element tables using the **ETABLE** command. Three strain components are considered, i.e., meridional, through-the-thickness, and hoop strains. Finally, total strains are obtained by adding elastic and plastic strains for each component using the **SADD** command, which adds columns to the element table. The element tables are listed in the *Output Window* using the **PRETAB** command. Figure 10.7 shows the listing of total strains at elements as they appear in the *Output Window*.

```
/POST1                          ! ENTER POSTPROCESSOR
RSYS,2                          ! ACTIVATE GLOBAL SPHERICAL
                                ! COORDINATE SYSTEM
ETABLE,EPELX,EPEL,X             ! STORE ELASTIC
                                ! STRAINS IN ELEMENT
ETABLE,EPELY,EPEL,Y             ! TABLE
ETABLE,EPELZ,EPEL,Z             !
ETABLE,EPPLX,EPPL,X             ! STORE PLASTIC
                                ! STRAINS IN ELEMENT
ETABLE,EPPLY,EPPL,Y             ! TABLE
ETABLE,EPPLZ,EPPL,Z             !
```

```
SADD,T_THK_TP,EPELX,EPPLX,1,1          ! ADD ELEMENT TABLE
SADD,T_MER_TP,EPELY,EPPLY,1,1          ! ITEMS TO FIND TOTAL
SADD,T_HOP_TP,EPELZ,EPPLZ,1,1          ! STRAINS
PRETAB,EPELX,EPELY,EPELZ               ! LIST ELASTIC
                                       ! STRAINS
PRETAB,EPPLX,EPPLY,EPPLZ               ! LIST PLASTIC
                                       ! STRAINS
PRETAB,T_MER_TP,T_THK_TP,T_HOP_TP      ! LIST TOTAL STRAINS
```

10.2.2 Plastic Deformation of an Aluminum Cylinder

Consider a thin-walled aluminum cylinder with a radius of $r = 20$ in., height of $r = 72$ in., a thickness of $t = 0.5$ in., and an extremely stiff panel on one end, as shown in Fig. 10.8. The cylinder and the stiff panel are first subjected to an internal pressure of $p_0 = 1500$ psi. With the internal pressure in place, the stiff panel is subjected to four tangential forces of 10^5 lb, as illustrated in Fig. 10.8. All displacements and rotations at the cylinder's end opposite to the stiff panel are constrained. The elastic modulus and Poisson's ratio of the aluminum shell are $E_{al} = 10^7$ psi and

Table 10.1 Data points for nonlinear stress-strain behavior of aluminum

ε	σ_e (psi)
0	0
0.00300	30,000
0.00350	33,041
0.00400	34,300
0.00450	35,267
0.00500	36,082
0.00550	36,800
0.00600	37,449
0.00650	38,045
0.00700	38,601
0.00750	39,123
0.00800	39,616
0.00850	40,086
0.00875	40,312
0.00900	40,534
0.00925	40,751
0.00950	40,964
0.00975	41,173
0.02400	49,708
0.04000	56,160
0.06300	63,313



```
PRETAB Command
File

PRINT ELEMENT TABLE ITEMS PER ELEMENT

 ***** POST1 ELEMENT TABLE LISTING *****

   STAT    CURRENT     CURRENT     CURRENT
   ELEM    T_MER_TP    T_THK_TP    T_HOP_TP
      1  0.49665E-02-0.67483E-02 0.49786E-02
      2  0.49665E-02-0.67483E-02 0.49785E-02
      3  0.49667E-02-0.67483E-02 0.49783E-02
      4  0.49669E-02-0.67482E-02 0.49781E-02
      5  0.49672E-02-0.67482E-02 0.49777E-02
      6  0.49677E-02-0.67482E-02 0.49773E-02
      7  0.49683E-02-0.67482E-02 0.49766E-02
      8  0.49693E-02-0.67484E-02 0.49758E-02
      9  0.49707E-02-0.67488E-02 0.49749E-02
     10  0.49722E-02-0.67494E-02 0.49739E-02
     11  0.49733E-02-0.67492E-02 0.49727E-02
     12  0.49731E-02-0.67469E-02 0.49704E-02
     13  0.49716E-02-0.67407E-02 0.49653E-02
     14  0.49736E-02-0.67323E-02 0.49542E-02
     15  0.49952E-02-0.67337E-02 0.49340E-02
     16  0.50695E-02-0.67762E-02 0.49054E-02
     17  0.52412E-02-0.69081E-02 0.48750E-02
     18  0.55275E-02-0.71471E-02 0.48435E-02
     19  0.57815E-02-0.72856E-02 0.47342E-02
     20  0.55931E-02-0.62723E-02 0.38121E-02

MINIMUM VALUES
ELEM            1           19           20
VALUE   0.49665E-02-0.72856E-02 0.38121E-02

MAXIMUM VALUES
ELEM           19           20            1
VALUE   0.57815E-02-0.62723E-02 0.49786E-02
```

Fig. 10.7 Listing of element table items as they appear in the *Output Window*



Fig. 10.8 Geometry (*top*) and the corresponding mesh (*bottom*) of the aluminum cylinder and the stiff panel

$\nu_{al} = 0.3$, respectively. The same set of properties corresponding to the stiff panel are $E_{st} = 10^{11}$ psi and $\nu_{st} = 0$. The plastic behavior of aluminum is governed by

Equation (10.3). The data points for the stress vs. strain are plotted in Fig. 10.6 and tabulated in Table 10.1. The goal is to obtain the plastic strain field resulting from the internal pressure and added torsion.

Model Generation

Both the aluminum cylinder and the stiff panel are modeled using an 8-noded shell element, **SHELL281**. For the nonlinear material behavior, the multiple-point isotropic hardening rule is chosen (**TB** command with **MISO** option). Although the geometry is axisymmetric, the loading is not. Therefore, the entire geometry is modeled.

```
/PREP7                  ! ENTER PREPROCESSOR
R=20                    ! PARAMETER FOR RADIUS
H=72                    ! PARAMETER FOR HEIGHT
T=R/40                  ! PARAMETER FOR THICKNESS
NDIV1=10                ! # OF DIVISIONS IN RADIAL
                        ! DIRECTION
NDIV2=15                ! # OF DIVISIONS IN HEIGHT
                        ! DIRECTION
P=1500                  ! INTERNAL PRESSURE
F=100E3                 ! TANGENTIAL FORCE
ET,1,281                ! USE SHELL281 ELEMENT TYPE
KEYOPT,1,4,0            ! NO USER SUBROUTINE FOR ELEMENT CS
KEYOPT,1,8,2            ! STORE DATA FOR TOP, BOTTOM & MID
                        ! SURFACES
MP,EX,1,1E7             ! SPECIFY ELASTIC MODULUS FOR
                        ! ALUMINUM
MP,NUXY,1,0.3           ! SPECIFY POISSON'S RATIO FOR
                        ! ALUMINUM
SECT,1,SHELL            ! SPECIFY SECTION TYPE
SECDATA,T,1             ! SPECIFY THICKNESS FOR ALUMINUM
TB,MISO,1,1,20,         ! MULTILINEAR ISOTROPIC HARDENING
                        ! PLASTICITY
TBPT,,0.00300,30000     ! ENTER STRAIN VS STRESS DATA
                        ! POINTS
TBPT,,0.00350,33041     !
TBPT,,0.00400,34300     !
TBPT,,0.00450,35267     !
TBPT,,0.00500,36082     !
TBPT,,0.00550,36800     !
TBPT,,0.00600,37449     !
TBPT,,0.00650,38045     !
TBPT,,0.00700,38601     !
TBPT,,0.00750,39123     !
TBPT,,0.00800,39616     !
TBPT,,0.00850,40086     !
TBPT,,0.00875,40312     !
TBPT,,0.00900,40534     !
TBPT,,0.00925,40751     !
TBPT,,0.00950,40964     !
TBPT,,0.00975,41173     !
TBPT,,0.02400,49708     !
```

```
TBPT,,0.04000,56160     !
TBPT,,0.06300,63313     !
MP,EX,2,1E11            !SPECIFY ELASTIC MODULUS FOR STIFF
                        !PANEL
MP,NUXY,2,0             !SPECIFY POISSON'S RATIO FOR STIFF
                        !PANEL
SECT,2,SHELL            !SPECIFY SECTION TYPE
SECDATA,T,2             !SPECIFY THICKNESS FOR STIFF PANEL
CYLIND,R,0,0,H          !CREATE CYLINDRICAL VOLUME
VDEL,ALL                !DELETE VOLUME (WITHOUT DELETING
                        !AREAS)
ASEL,S,LOC,Z,0          !SELECT AREAS AT Z = 0
ASEL,A,LOC,Z,H          !ADD AREAS AT Z = H TO SELECTION
ADEL,ALL                !DELETE SELECTED AREAS
ALLSEL                  !SELECT EVERYTHING
LSEL,S,LOC,Z,H/2        !SELECT LINES AT Z = H/2
LESIZE,ALL,,,NDIV2      !SPECIFY # OF DIVISIONS ON
                        !SELECTED LINES
ALLSEL                  !SELECT EVERYTHING
LESIZE,ALL,,,NDIV1      !SPECIFY # OF DIVISIONS ON ALL
                        !LINES
LSEL,S,LOC,Z,0          !SELECT LINES AT Z = 0
LSEL,R,LOC,Y,0,R        !RESELECT LINES BETWEEN Y = 0 AND
                        !y = R
LCCAT,ALL               !CONCATENATE LINES FOR MAPPED
                        !MESHING
LSEL,S,LOC,Z,0          !SELECT LINES AT Z = 0
LSEL,R,LOC,Y,0,-R       !RESELECT LINES BETWEEN Y = 0 AND
                        !Y = -R
LCCAT,ALL               !CONCATENATE LINES FOR MAPPED
                        !MESHING
LSEL,S,LOC,Z,H          !SELECT LINES AT Z = H
LSEL,R,LOC,Y,0,R        !RESELECT LINES BETWEEN Y = 0 AND
                        !Y = R
LCCAT,ALL               !CONCATENATE LINES FOR MAPPED
                        !MESHING
LSEL,S,LOC,Z,H          !SELECT LINES AT Z = H
LSEL,R,LOC,Y,0,-R       !RESELECT LINES BETWEEN Y = 0 AND
                        !Y = -R
LCCAT,ALL               !CONCATENATE LINES FOR MAPPED
                        !MESHING
ALLSEL                  !SELECT EVERYTHING
MSHKEY,1                !ENFORCE MAPPED MESHING
MAT,1                   !SWITCH MATERIAL ATTRIBUTE TO
                        !MAT # 1
SECNUM,1                !SWITCH SECTION ATTRIBUTE TO
                        !SEC # 1
AMESH,ALL               !MESH ALL AREAS
WPOFFS,0,0,H            !OFFSET WORKING PLANE
PCIRC,R,0,0,90          !CREATE CIRCLE
PCIRC,1.5*R,R,0,90      !CREATE HOLLOW CIRCLE
ASEL,S,LOC,Z,H          !SELECT AREAS AT Z = H
AGLUE,ALL               !GLUE SELECTED AREAS
CSYS,1                  !SWITCH TO GLOBAL CYLINDRICAL CS
```

```
LSLA,S                   !SELECT LINES ATTACHED TO SELECTED
                         !AREAS
LSEL,U,LOC,X,1.25*R      !UNSELECT LINES AT X = 1.25*R
LESIZE,ALL,,,NDIV1       !SPECIFY # OF DIVS ON SELECTED
                         !LINES
LSLA,S                   !SELECT LINES ATTACHED TO SELECTED
                         !AREAS
LSEL,R,LOC,X,1.25*R      !RESELECT LINES AT X = 1.25*R
LESIZE,ALL,,,NDIV1/2     !SPECIFY # OF DIVS ON SELECTED
                         !LINES
MAT,2                    !SWITCH MATERIAL ATTRIBUTE TO MAT
                         !# 2
SECNUM,2                 !SWITCH SECTION ATTRIBUTE TO SEC
                         !# 2
AMESH,ALL                !MESH SELECTED AREAS
CSYS                     !SWITCH TO GLOBAL CARTESIAN CS
ARSYM,X,ALL              !REFLECT SELECTED AREAS ABOUT Y-Z
                         !PLANE
ARSYM,Y,ALL              !REFLECT SELECTED AREAS ABOUT X-Z
                         !PLANE
ALLSEL                   !SELECT EVERYTHING
NUMMRG,ALL               !MERGE DUPLICATE ENTITIES
FINISH                   !EXIT PREPROCESSOR
```

Solution

The solution is obtained in two steps. First, the internal pressure is applied and the first load step file is generated (**LSWRITE** command). Then, the concentrated loads are applied while the internal pressure is still present, which constitutes the second load step file. The solution is obtained sequentially from these load step files using the **LSSOLVE** command.

```
/SOLU                    !ENTER SOLUTION PROCESSOR
NSEL,S,LOC,Z,0           !SELECT NODES AT Z = 0
D,ALL,ALL                !CONSTRAIN ALL DOFS (DISPL &
                         !ROTATIONS)
ALLSEL                   !SELECT EVERYTHING
ESEL,S,MAT,,1            !SELECT ELEMENTS WITH MAT # 1
SFE,ALL,1,PRES,,P        !APPLY INTERNAL PRES. ON SELECTED
                         !ELEMS
CSYS,1                   !SWITCH TO GLOBAL CYLINDRICAL CS
ASEL,S,LOC,Z,H           !SELECT AREAS AT Z = H
ASEL,R,LOC,X,-R,R        !RESELECT AREAS BETWEEN X = -R &
                         !X = R
CSYS                     !SWITCH TO GLOBAL CARTESIAN CS
ESLA,S                   !SELECT ELEMENTS ATTACHED TO
                         !SELECTED AREAS
SFE,ALL,1,PRES,,P        !APPLY INTERNAL PRES. ON SELECTED
                         !ELEMS
ALLSEL                   !SELECT EVERYTHING
LSWRITE,1                !WRITE LOAD STEP FILE # 1
N=NODE(1.5*R,0,H)        !STORE NODE # IN N FOR GIVEN
                         !COORDS.
```

```
F,N,FY,F                ! APPLY CONCENTRATED LOAD ON NODE N
N=NODE(-1.5*R,0,H)      ! STORE NODE # IN N FOR GIVEN
                        ! COORDS.
F,N,FY,-F               ! APPLY CONCENTRATED LOAD ON NODE N
N=NODE(0,1.5*R,H)       ! STORE NODE # IN N FOR GIVEN
                        ! COORDS.
F,N,FX,-F               ! APPLY CONCENTRATED LOAD ON NODE N
N=NODE(0,-1.5*R,H)      ! STORE NODE # IN N FOR GIVEN
                        ! COORDS.
F,N,FX,F                ! APPLY CONCENTRATED LOAD ON NODE N
ALLSEL                  ! SELECT EVERYTHING
LSWRITE,2               ! WRITE LOAD STEP FILE # 1
LSSOLVE,1,2             ! OBTAIN SOLUTION FROM LOAD STEPS
                        ! FILES
FINISH                  ! EXIT SOLUTION PROCESSOR
```

Postprocessing

Once the solution is complete, the quantities of interest, i.e., radial, circumferential, and longitudinal displacements and circumferential plastic strains, are retrieved at a specific node and written to an external text file for both load steps. The channeling of data to the external text file, referred to as the *Command File*, is achieved by using the ***CFOPEN** command (detailed description is given in Sect. 11.5.2.2). After issuing the ***CFOPEN** command, quantities of interest are written to the file **PLASTIC OUT** using the ***VWRITE** command. The aforementioned quantities are reviewed at a single node, which is chosen along the middle of the cylinder surface in the z-direction (i.e., $z = H/2$). Plastic strains are calculated for elements. Therefore, plastic strains in the circumferential direction are stored in an element table using the **ETABLE** command. There are two elements attached to the selected single node. Therefore, the plastic strains at these two elements are averaged to find its value at the shared node. The results associated with specific load steps are read using the **SET** command.

```
/POST1                     ! ENTER GENERAL POSTPROCESSOR
*CFOPEN,PLASTIC,OUT        ! OPEN “COMMAND FILE”
CSYS,1                     ! SWITCH TO GLOBAL CYLINDRICAL
                           ! CS
RSYS,1                     ! USE CYLINCRICAL CS FOR
                           ! RESULTS
SET,1                      ! READ RESULTS OF LOAD STEP 1
ETABLE,EPT,EPPL,Y          ! STORE PLASTIC STRAINS IN
                           ! THETA DIRECTION IN ELEMENT
                           ! TABLE
NSEL,S,LOC,Z,H/2           ! SELECT NODES AT Z = H/2
NSEL,R,LOC,Y,0             ! RESELECT NODES AT THETA = 0
ESLN,S                     ! SELECT ELEMENTS ATTACHED TO
                           ! SELECTED NODE
*GET,E1,ELEM,0,NUM,MAX     ! STORE MAX ELEM # IN PARAMETER
                           ! E1
*GET,E2,ELEM,0,NUM,MIN     ! STORE MIN ELEM # IN PARAMETER
                           ! E2
```

```
*GET,EPT1,ETAB,1,ELEM,E1    ! STORE ELEM TABLE ITEMS FOR E1
*GET,EPT2,ETAB,1,ELEM,E2    ! & E2 IN PARAMETERS EPT1 &
                            ! EPT2
EPTAV=(EPT1+EPT2)/2         ! FIND AVERAGE OF EPT1 & EPT2
*GET,NODE1,NODE,0,NUM,MAX   ! RETRIEVE SELECTED NODE NUMBER
*GET,UR1,NODE,NODE1,U,X     ! RETRIEVE RADIAL DISPLACEMENT
*GET,UT1,NODE,NODE1,U,Y     ! RETRIEVE CIRCUMF.
                            ! DISPLACEMENT
*GET,UZ1,NODE,NODE1,U,Z     ! RETRIEVE Z-DISPLACEMENT
*VWRITE,'TIME = 1'          ! START WRITING TO COMMAND FILE
(A8)                        ! FORMAT STATEMENT
*VWRITE,'UR = ',UR1         ! WRITE RADIAL DISPLACEMENT
(A9,E14.5)                  ! FORMAT STATEMENT
*VWRITE,'UTHETA = ',UT1     ! WRITE CIRCUMF. DISPLACEMENT
(A9,E14.5)                  ! FORMAT STATEMENT
*VWRITE,'UZ = ',UZ1         ! WRITE Z-DISPLACEMENT
(A9,E14.5)                  ! FORMAT STATEMENT
*VWRITE,'EPLTH = ',EPTAV    ! WRITE AVERAGED PLASTIC STRAIN
(A9,E14.5)                  ! FORMAT STATEMENT
SET,NEXT                    ! READ RESULTS OF LOAD STEP 2
ETABLE,REFL                 ! UPDATE ELEMENT TABLE
*GET,EPT1,ETAB,1,ELEM,E1    ! STORE ELEM TABLE ITEMS FOR E1
*GET,EPT2,ETAB,1,ELEM,E2    ! & E2 IN PARAMETERS EPT1 &
                            ! EPT2
EPTAV=(EPT1+EPT2)/2         ! FIND AVERAGE OF EPT1 & EPT2
*GET,UR1,NODE,NODE1,U,X     ! RETRIEVE RADIAL DISPLACEMENT
*GET,UT1,NODE,NODE1,U,Y     ! RETRIEVE CIRCUMF.
                            ! DISPLACEMENT
*GET,UZ1,NODE,NODE1,U,Z     ! RETRIEVE Z-DISPLACEMENT
*VWRITE,'TIME = 2'          ! RESUME WRITING TO COMMAND
                            ! FILE
(A8)                        ! FORMAT STATEMENT
*VWRITE,'UR = ',UR1         ! WRITE RADIAL DISPLACEMENT
(A9,E14.5)                  ! FORMAT STATEMENT
*VWRITE,'UTHETA = ',UT1     ! WRITE CIRCUMF. DISPLACEMENT
(A9,E14.5)                  ! FORMAT STATEMENT
*VWRITE,'UZ = ',UZ1         ! WRITE Z-DISPLACEMENT
(A9,E14.5)                  ! FORMAT STATEMENT
*VWRITE,'EPLTH = ',EPTAV    ! WRITE AVERAGED PLASTIC STRAIN
(A9,E14.5)                  ! FORMAT STATEMENT
*CFCLOS                     ! CLOSE "COMMAND FILE"
FINISH                      ! EXIT GENERAL POSTPROCESSOR
```

After the command input segment given above is executed, the contents of the file PLASTIC OUT is given as

```
TIME = 1
 UR     =   0.52429E+00
 UTHETA =   0.46792E-09
 UZ     =   0.60653E-01
 EPLTH  =   0.20618E-01
TIME = 2
 UR     =   0.62531E+00
 UTHETA =   0.19920E+00
 UZ     =   0.64957E-01
 EPLTH  =   0.25581E-01
```

10.2.3 Stress Analysis of a Reinforced Viscoelastic Cylinder

Consider a hollow viscoelastic cylinder reinforced by an elastic material along the outer periphery subjected to internal pressure, as shown in Fig. 10.9. The cylinder is long in the out-of-plane direction. The inner and outer radii of the cylinder are $a = 2$ in. and $b = 4$ in., respectively; the thickness of the reinforcing layer is $h = 4/33$ in. The elastic modulus and Poisson's ratio of the reinforcement are $E_r = 3 \times 10^7$ psi and $\nu_r = 1/\sqrt{11}$.

The shear and bulk moduli of the viscoelastic cylinder behave as

$$G(t) = G_0 e^{-t} \tag{10.4a}$$

$$K(t) = K_\infty H(t) \tag{10.4b}$$

in which G_0 and K_∞ are defined as $G(0) = G_0 = E_c / (1 + 2\nu_c)$ and $K_\infty = E_c / [3(1 - 2\nu_c)]$. The elastic modulus and Poisson's ratio of the cylinder are $E_c = 10^5\ psi$ and $\nu_c = 1/3$.

Within ANSYS, the viscoelastic material behavior is specified as

$$G(t) = G_0 \left[\frac{G_\infty}{G_0} + \frac{G_1}{G_0} e^{\left(-t/\tau^G\right)} \right] \tag{10.5a}$$

$$K(t) = K_0 \left[\frac{K_\infty}{K_0} + \frac{K_1}{K_0} e^{\left(-t/\tau^K\right)} \right] \tag{10.5b}$$

In accordance with Eq. (10.4), the parameters in Eq. (10.5) take the following values: $G(0) = G_0 = G_1$, $\lim_{t \to \infty} G(t) = G_\infty = 0$, $K_0 = K_\infty$, and $K_1 = 0$. Therefore, the relative shear and bulk moduli, (G_1 / G_0) and (K_1 / K_0), respectively, are assigned the values of 1.0 and 0.0. The parameters τ^G and τ^K represent relative times for shear and bulk behavior, respectively; both of them are assumed to be 1.0. The goal is to find the time-dependent behavior of radial and hoop stresses.

Model Generation

The problem is solved using two-dimensional 8-noded **PLANE183** elements with the axisymmetric option. A Prony series representation of viscoelasticity is specified **TB** using and **TBDATA** commands. The model is a rectangular cross section of the cylinder and the reinforcement layer with a height of 1 in.

```
/PREP7                  !ENTER PREPROCESSOR
ET,1,PLANE183,,,1       !USE PLANE183 ELEM WITH
                        !AXISYMMETRY
MP,EX,1,1.0E5           !ELASTIC MODULUS OF MAT 1
MP,NUXY,1,1/3           !POISSON'S RATIO OF MAT 1
TB,PRONY,1,,1,SHEAR     !USE PRONY SERIES VISCOELASTICITY
                        !FOR SHEAR
TBDATA,1,1.0,1.0        !RELATIVE MODULUS AND TIME
TB,PRONY,1,,1,BULK      !USE PRONY SERIES VISCOELASTICITY
                        !FOR BULK
TBDATA,1,0.0,1.0        !RELATIVE MODULUS AND TIME
MP,EX,2,3.0E7           !ELASTIC MODULUS OF MAT 2
MP,NUXY,2,1/SQRT(11)    !POISSON'S RATIO OF MAT 2
R1=2                    !PARAMETER FOR INNER RADIUS OF
                        !CYLINDER
R2=4                    !PARAMETER FOR OUTER RADIUS OF
                        !CYLINDER
R3=R2+4/33              !PARAMETER FOR OUTER RADIUS OF
                        !REINFORC.
K,1,R1                  !CREATE KEYPOINTS
K,2,R2                  !
K,3,R3                  !
KGEN,2,1,3,1,,1         !GENERATE KPS FROM EXISTING
                        !PATTERN
L,1,2,5                 !CREATE LINES
L,2,3,1                 !
A,4,1,2,5               !CREATE AREAS
A,5,2,3,6               !
LESIZE,1,,,10           !SPECIFY # OF LINE DIVISIONS
LESIZE,2,,,2            !
ESIZE,0.5               !SPECIFY ELEMENT SIZE
MSHKEY,1                !ENFORCE MAPPED MESHING
MSHAPE,0,2D             !ENFORCE QUADRILATERAL ELEMENT
                        !SHAPE
MAT,1                   !SWITCH TO MATERIAL 1
```



Fig. 10.9 Hollow viscoelastic cylinder with external reinforcement

```
AMESH,1                  !MESH AREA 1
MAT,2                    !SWITCH TO MATERIAL 2
AMESH,2                  !MESH AREA 2
FINISH                   !EXIT PREPROCESSOR
```

Solution

The solution is obtained in two load steps. The first load step, which spans an extremely small duration of 0.00001 s, is used to set up the initial conditions for the problem. As shown in Fig. 10.9, the nodes along the bottom row are constrained to move in the vertical direction (*y*-direction). In order to capture the plane strain characteristic of the problem, displacements are enforced to be the same on the top row of nodes. This is accomplished by coupling the degree of freedom in the *y*-direction of those nodes using the **CP** command. The solution controls option is turned off in this analysis, and values of the specific nonlinear solution controls items are specified. In order to reduce the size of the results file, only a limited amount of data is written (**OUTRES** command). The accuracy of the solution is improved by specifying a small convergence tolerance value using the **CNVTOL** command. Finally, the time step size is specified to be 0.1 s using the **DELTIM** command.

```
/SOLU                     ! ENTER SOLUTION PROCESSOR
ANTYPE,STATIC             ! SPECIFY ANALYSIS TYPE
NSEL,S,LOC,Y,0            ! SELECT NODES AT Y = 0
D,ALL,UY                  ! CONSTRAIN Y-DISP AT SELECTED
                          ! NODES
NSEL,S,LOC,Y,1            ! SELECT NODES AT Y = 1
CP,1,UY,ALL               ! COUPLE Y-DISP OF SELECTED NODES
ALLSEL                    ! SELECT EVERYTHING
NSEL,S,LOC,X,R1           ! SELECT NODES AT X = R1
PI0 = 1                   ! PARAMETER FOR INNER PRESSURE
SF,,PRES,PI0              ! APPLY PRESSURE BC AT SELECTED
                          ! NODES
ALLSEL                    ! SELECT EVERYTHING
SOLCONTROL,0              ! TURN SOLUTION CONTROLS OFF
OUTRES,BASIC,ALL          ! SAVE BASIC OUTPUT AT EVERY
                          ! SUBSTEP
CNVTOL,F,,,,1E-7          ! SMALL CONVERGENCE TOLER. ENFORCED
TIME,0.00001              ! TIME AT THE END OF 1ST LOAD STEP
SOLVE                     ! OBTAIN SOLUTION FOR 1ST LOAD STEP
TIME,10                   ! TIME AT THE END OF 2ND LOAD STEP
DELTIM,0.1                ! SPECIFY TIME STEP SIZE
SOLVE                     ! OBTAIN SOLUTION FOR 2ND LOAD STEP
```

Postprocessing
Once the solution is obtained, a component of nodes (the center row of nodes) is created using the **CM** command. The radial and hoop stresses are written to an external file, which is opened using the ***CFOPEN** command and closed using the ***CFCLOSE** command. Between these two commands, desired quantities are written to the external file using the ***VWRITE** command followed by a format



Fig. 10.10 Variation of radial stress (σ_{rr}) along $y = 0.5$ in at different times

statement (written in FORTRAN syntax). Figs. 10.10 and 10.11 show the variation of radial (σ_{rr}) and hoop ($\sigma_{\theta\theta}$) stresses, respectively, along $y = 0.5$ in for times $t = 0.1, 1, 5,$ and 10.

```
/POST1                          ! ENTER POSTPROCESSOR
ESEL,S,MAT,,1                   ! SELECT ELEMENTS WITH MAT 1
NSLE,S,CORNER                   ! SELECT CORNER NODES ATTACHED
                                ! TO THE SELECTED ELEMENTS
NSEL,R,LOC,Y,0.5                ! RESELECT NODES AT Y = 0.5
CM,NLIST,NODE                   ! CREATE A COMPONENT OF
                                ! NODES NAMED "NLIST"
*CFOPEN,'STRS','OUT'            ! OPEN DATA FILE "STRS.OUT"
SET,FIRST                       ! READ RESULTS FOR 1ST LOAD
                                ! STEP
*DO,J,1,1000                    ! LOOP OVER RESULTS SETS
ALLSEL                          ! SELECT EVERYTHING
*GET,TIM,ACTIVE,0,SET,TIME      ! OBTAIN TIME AT CURRENT
                                ! RESULTS SET
*VWRITE,'********'              ! WRITE A SEPARATOR ROW TO
                                ! FILE
(A8)                            ! FORMAT STATEMENT
*VWRITE,'TIME = ',TIM           ! WRITE CURRENT TIME TO FILE
(A7,E10.3)                      ! FORMAT STATEMENT
NSEL,S,NODE,,NLIST              ! SELECT COMPONENT "NLIST"
*GET,NCOUNT,NODE,0,COUNT        ! OBTAIN NUMBER OF NODES
*DO,I,1,NCOUNT,1                ! LOOP OVER SELECTED NODES
NODNUM = NODE(0.0,0.0,0.0)      ! OBTAIN NODE # CLOSEST
                                ! TO THE CENTER
LOCA=NX(NODNUM)                 ! OBTAIN X-COORD OF
                                ! THE NODE "NODNUM"
*GET,SIGRR,NODE,NODNUM,S,X      ! OBTAIN σrr AT NODE "NODNUM"
*GET,SIGTT,NODE,NODNUM,S,Z      ! OBTAIN σθθ AT NODE "NODNUM"
*VWRITE,LOCA,SIGRR,SIGTT        ! WRITE σrr AND σθθ TO FILE
(E10.3,3X,E10.3,3X,E10.3)       ! FORMAT STATEMENT
NSEL,U,NODE,,NODNUM             ! UNSELECT THE NODE "NODNUM"
*ENDDO                          ! END LOOP OVER NODES
*IF,TIM,GE,10.0,*EXIT           ! IF TIM = 10, END LOOP
SET,NEXT                        ! READ NEXT RESULTS SET
*ENDDO                          ! END LOOP OVER RESULTS SETS
*CFCLOSE                        ! CLOSE DATA FILE
ALLSEL                          ! SELECT EVERYTHING
```

10.2.4 Viscoplasticity Analysis of a Eutectic Solder Cylinder

Consider a cylindrical eutectic solder with a radius of 10 mm and a height of 100 mm, as shown in Fig. 10.12. The bottom surface of the cylinder is constrained



Fig. 10.11 Variation of hoop stresses ($\sigma_{\theta\theta}$) along $y = 0.5$ in at different times

Fig. 10.12 Cylindrical eutectic solder



in all directions while the top surface is subjected to a prescribed displacement in the y-direction as a sinusoidal function of time. In addition, the eutectic solder is exposed to temperature, which exhibits the same time-dependent behavior as the prescribed displacement.

These loading conditions are given by

$$u_y(x, y = 100, z, t) = u_y^{\max} \sin\left(\frac{\pi t}{t_{fin}}\right)$$

$$T(x, y, z, t) = T_{\min} + (T_{\max} - T_{\min}) \sin\left(\frac{\pi t}{t_{fin}}\right) \tag{10.6}$$

Table 10.2 Temperature-dependent variation of material properties of the eutectic solder

Temperature (°C)	Elastic modulus (MPa)	Poisson's ratio	Coefficient of thermal expansion (ppm/°C)
−35	40,781	0.3540	24.27
−15	37,825	0.3565	24.48
5	34,884	0.3600	24.66
25	31,910	0.3628	24.80
50	28,149	0.3650	25.01
75	24,425	0.3700	25.26
100	20,710	0.3774	25.52
125	16,942	0.3839	25.79

in which u_y^{max} and T_{max} are the maximum values of the applied displacement and temperature, respectively; T_{min} is the stress-free temperature; t designates time; and t_{fin} is the time at the end of the process. The numerical values of these parameters are given as $u_y^{max} = 0.5\ \text{mm}$, $T_{min} = 0°\text{C}$, $T_{max} = 125°\text{C}$, and $t_{fin} = 60\text{s}$. The material properties of the solder vary with temperature, as tabulated in Table 10.2. In addition to the elastic material properties, Anand's viscoplastic material behavior is assumed for the solder, with related parameters listed in Table 10.3. The goal is to find strain field (elastic, inelastic and total strains) at different times.

Model Generation

The problem is solved using **`PLANE182`** element with the axisymmetric option. In order to specify temperature-dependent elastic properties, a temperature table is constructed using the **`MPTEMP`** command, followed by the specification of properties using the **`MPDATA`** command. Note that temperatures are specified in Kelvin. Anand's viscoplastic properties are specified using the **`TB`** and **`TBDATA`** commands. In the analysis, millimeter (mm) is used for length dimensions and megapascal (MPa) is used for stresses (also elastic modulus). Thus, the resulting displacements and stresses are in millimeters and megapascals, respectively. However, this is a special case where an inconsistent unit system works correctly, and it is highly recommended that a unit analysis be performed before applying similar approaches to other mixed-unit systems. The plastic behavior at different temperatures is shown in Fig. 10.13.

Table 10.3 Numerical values of parameters used in Anand's material model for the eutectic solder

Parameter	Description	Value
S_0	Initial deformation resistance	12.41 (MPa)
Q/R	Ratio of activation energy to universal gas constant	9400 (1/°K)
A	Pre-exponential factor	$4\times10^6 (1/\text{sec})$
ξ	Stress multiplier	1.5
m	Strain rate sensitivity of stress	0.303
h_0	Hardening/softening constant	1379 (MPa)
$\hat{s}$	Coefficient for deformation resistance saturation value	13.79 (MPa)
n	Strain rate sensitivity of saturation value	0.07
a	Strain rate sensitivity of hardening or softening	1.3

Fig. 10.13 Nonlinear stress-strain behavior of solder at different temperatures




```
/PREP7                       !ENTER PREPROCESSOR
ET,1,PLANE182                !USE PLANE182 ELEMENT
KEYOPT,1,3,1                 !SPECIFY AXISYMMETRY
MPTEMP                       !INITIALIZE MATERIAL TABLE
                             !SPECIFY TEMPERATURE POINTS
MPTEMP,1,238.15,258.15,278.15
MPTEMP,4,298.15,323.15,348.15,373.15,398.15
                             !SPECIFY ELASTIC MODULUS
MPDATA,EX,1,1,40781,37825,34884
MPDATA,EX,1,4,31910,28149,24425,20710,16942
                             !SPECIFY POISSON'S RATIO
MPDATA,NUXY,1,1,0.354,0.3565,0.36
MPDATA,NUXY,1,4,0.3628,0.365,0.37,0.3774,0.3839
                             !SPECIFY COEFFICIENT OF THERMAL
                             !EXPANSION
MPDATA,ALPX,1,1,2.427e-5,2.448e-5,2.466e-5
MPDATA,ALPX,1,4,2.48e-5,2.501e-5,2.526e-5,2.552e-5,
                                                      ➥2.579e-5
TB,RATE,1,1,9,9              !SPECIFY ANAND'S VISCOPLASTICITY
                             !SPECIFY ANAND'S PARAMETERS
TBDATA,,12.41,9400,4e6,1.5,0.303,1379
TBDATA,,13.79,0.07,1.3
TB,MISO,1,8,7,                  !SPECIFY ISOTROPIC HARDENING
TBTEMP,-35+273.15               !SPECIFY TEMPERATURE AS -35 C
TBPT,DEFI,9.313161E-04,37.980     !SPECIFY STRAIN VS
TBPT,DEFI,8.931316E-03,50.677     !STRESS DATA POINTS AT
TBPT,DEFI,1.693132E-02,56.307     !THIS TEMPERATURE
TBPT,DEFI,2.493132E-02,60.142     !
TBPT,DEFI,4.893132E-02,67.620     !
TBPT,DEFI,8.093132E-02,73.889     !
TBPT,DEFI,1.009313E-01,76.836     !
TBTEMP,-15+273.15               !SPECIFY TEMPERATURE AS -15 C
TBPT,DEFI,9.295968E-04,35.162     !SPECIFY STRAIN VS
TBPT,DEFI,8.929597E-03,46.917     !STRESS DATA POINTS AT
```

```
TBPT,DEFI,1.692960E-02,52.129     !THIS TEMPERATURE
TBPT,DEFI,2.492960E-02,55.680     !
TBPT,DEFI,4.892960E-02,62.603     !
TBPT,DEFI,8.092960E-02,68.407     !
TBPT,DEFI,1.009296E-01,71.136     !
TBTEMP,5+273.15                !SPECIFY TEMPERATURE AS 5 C
TBPT,DEFI,9.271872E-04,32.344     !SPECIFY STRAIN VS
TBPT,DEFI,8.927187E-03,43.158     !STRESS DATA POINTS AT
TBPT,DEFI,1.692719E-02,47.952     !THIS TEMPERATURE
TBPT,DEFI,2.492719E-02,51.218     !
TBPT,DEFI,4.892719E-02,57.586     !
TBPT,DEFI,8.092719E-02,62.925     !
TBPT,DEFI,1.009272E-01,65.435     !
TBTEMP,25+273.15               !SPECIFY TEMPERATURE AS 25 C
TBPT,DEFI,9.252899E-04,29.526     !SPECIFY STRAIN VS
TBPT,DEFI,8.925290E-03,39.398     !STRESS DATA POINTS AT
TBPT,DEFI,1.692529E-02,43.774     !THIS TEMPERATURE
TBPT,DEFI,2.492529E-02,46.756     !
TBPT,DEFI,4.892529E-02,52.569     !
TBPT,DEFI,8.092529E-02,57.443     !
TBPT,DEFI,1.009253E-01,59.734     !
TBTEMP,50+273.15               !SPECIFY TEMPERATURE AS 50 C
TBPT,DEFI,9.237984E-04,26.004     !SPECIFY STRAIN VS
TBPT,DEFI,8.923798E-03,34.698     !STRESS DATA POINTS AT
TBPT,DEFI,1.692380E-02,38.552     !THIS TEMPERATURE
TBPT,DEFI,2.492380E-02,41.178     !
TBPT,DEFI,4.892380E-02,46.298     !
TBPT,DEFI,8.092380E-02,50.590     !
TBPT,DEFI,1.009238E-01,52.608     !
TBTEMP,75+273.15               !SPECIFY TEMPERATURE AS 75 C
TBPT,DEFI,9.204504E-04,22.482     !SPECIFY STRAIN VS
TBPT,DEFI,8.920450E-03,29.998     !STRESS DATA POINTS AT
TBPT,DEFI,1.692045E-02,33.330     !THIS TEMPERATURE
TBPT,DEFI,2.492045E-02,35.601     !
TBPT,DEFI,4.892045E-02,40.027     !
TBPT,DEFI,8.092045E-02,43.738     !
TBPT,DEFI,1.009205E-01,45.482     !
TBTEMP,100+273.15              !SPECIFY TEMPERATURE AS 100 C
TBPT,DEFI,9.154515E-04,18.959     !SPECIFY STRAIN VS
TBPT,DEFI,8.915451E-03,25.298     !STRESS DATA POINTS AT
TBPT,DEFI,1.691545E-02,28.108     !THIS TEMPERATURE
TBPT,DEFI,2.491545E-02,30.023     !
TBPT,DEFI,4.891545E-02,33.756     !
TBPT,DEFI,8.091545E-02,36.885     !
TBPT,DEFI,1.009155E-01,38.357     !
TBTEMP,125+273.15              !SPECIFY TEMPERATURE AS 125 C
TBPT,DEFI,9.111675E-04,15.437     !SPECIFY STRAIN VS
TBPT,DEFI,8.911168E-03,20.598     !STRESS DATA POINTS AT
TBPT,DEFI,1.691117E-02,22.886     !THIS TEMPERATURE
TBPT,DEFI,2.491117E-02,24.445     !
TBPT,DEFI,4.891117E-02,27.485     !
TBPT,DEFI,8.091117E-02,30.033     !
TBPT,DEFI,1.009112E-01,31.231     !
RECTNG,0,10,0,100           !CREATE RECTANGLE
ESIZE,2                     !SPECIFY ELEMENT SIZE
```

```
MSHKEY,1                    !ENFORCE MAPPED MESHING
AMESH,ALL                   !MESH AREA
FINISH                      !EXIT PREPROCESSOR
```

Solution

The solution is obtained in twenty load steps. Nonlinear geometry effects must be turned on when using **`PLANE182`** elements. Automatic time stepping is turned on (**`AUTOTS`** command) so that ANSYS can adjust the time step size values for the substeps within each load step. However, the starting time step size and the minimum and maximum time step size to be used in the analysis are specified using the **`DELTIM`** command. A do loop is utilized for creating load step files, in which the applied displacement and temperature conditions are calculated and written to load step files. Finally, the solution is obtained using the **`LSSOLVE`** command.

```
/SOLU                          !ENTER SOLUTION PROCESSOR
TMIN=273.15                    !PARAMETER FOR MIN
                               !TEMPERATURE
TMAX=398.15                    !PARAMETER FOR MAX
                               !TEMPERATURE
DELT=TMAX-TMIN                 !TEMPERATURE DIFFERENCE
VMAX=0.5                       !MAXIMUM APPLIED
                               !DISPLACEMENT
NLS=20                         !NUMBER OF LOAD STEPS
TMX=60                         !FINAL TIME (IN SECONDS)
PI=4*ATAN(1)                   !PARAMETER FOR π
ANTYPE,STATIC                  !DECLARE ANALYSIS TYPE
NLGEOM,ON                      !TURN NONLINEAR GEOMETRY ON
KBC,0                          !APPLY RAMPED LOADING
AUTOTS,ON                      !TURN AUTOMATIC TIME
                               !STEPPING ON
TREF,273.15                    !STRESS-FREE TEMPERATURE
DELTIM,1E-2,1E-3,0.3           !SPECIFY TIME STEP SIZE
NSEL,S,LOC,Y,0                 !SELECT NODES AT Y = 0
D,ALL,ALL                      !CONSTRAIN ALL DOFS
ALLSEL                         !SELECT EVERYTHING
TIM=TMX/NLS                    !FIND TIME AT CURRENT LOAD
                               !STEP
*DO,I,1,NLS                    !BEGIN DO LOOP ON LOAD STEPS
TIME,TIM                       !SPECIFY TIME AT END OF LOAD
                               !STEP
VC=VMAX*SIN(PI*TIM/TMX)        !FIND CURRENT DISPLACEMENT
                               !BC
TC=TMIN+DELT*SIN(PI*TIM/TMX)   !FIND CURRENT TEMPERATURE
NSEL,S,LOC,Y,100               !SELECT NODES AT Y = 100
D,ALL,UY,VC                    !APPLY CURRENT DISPLACEMENT
                               !BC
ALLSEL                         !SELECT EVERYTHING
BFUNIF,TEMP,TC                 !APPLY CURRENT TEMPERATURE
                               !LOAD
LSWRITE,I                      !WRITE LOAD STEP FILE
```

```
TIM=TIM+TMX/NLS                 !UPDATE TIME AT END OF LOAD
                                !STEP
*ENDDO                          !END DO LOOP ON LOAD STEPS
LSSOLVE,1,NLS                   !SOLVE FROM LOAD STEP FILES
FINISH                          !EXIT SOLUTION PROCESSOR
```

Postprocessing
Once the solution is obtained, the user can review a multitude of results items, including elastic, plastic, and total strain components and stress components. Also available is the plastic work. The command input segment given below obtains contour plots for a few of the items mentioned above, in two different load steps. The results for different load steps are read using the **SET** command. For clarity in the contour plots, the triad and the symbols for minimum and maximum quantities are turned off using the **/TRAID** and **/PLOPTS** commands. The two load steps considered here are the 10th and the last load steps. The 10th load step corresponds to the maximum displacement and temperature conditions while the last load step brings the structure to its initial configuration. Although initially stress free, after applying time-dependent displacement and temperature loads and returning to a no-load state, the structure experiences residual stresses and strains. Figs. 10.14 and 10.15 show the contour plots of plastic work at load steps 10 and 20, respectively.

```
/POST1                 !ENTER POSTPROCESSOR
/TRIAD,OFF             !TURN THE TRIAD OFF
/PLOPTS,MINM,0         !TURN MIN & MAX SYMBOLS OFF
SET,10                 !READ RESULTS AT LOAD STEP 10
PLNSOL,S,EQV           !PLOT EQUIVALENT STRESS CONTOURS
PLNSOL,EPPL,Y          !PLOT CONTOURS FOR PLASTIC STRAIN
                       !IN Y-DIR.
PLNSOL,NL,PLWK         !PLOT PLASTIC WORK CONTOURS
SET,20                 !READ RESULTS AT LOAD STEP 20
PLNSOL,S,EQV           !PLOT EQUIVALENT STRESS CONTOURS
PLNSOL,EPPL,Y          !PLOT CONTOURS FOR PLASTIC STRAIN
                       !IN Y-DIR.
PLNSOL,NL,PLWK         !PLOT PLASTIC WORK CONTOURS
```

10.2.5 Combined Plasticity and Creep

In this subsection, the cylindrical eutectic solder column considered in the previous subsection is reconsidered, this time with nonlinear material properties by means of rate-independent plasticity combined with creep (Fig. 10.12). Both top and bottom surfaces of the cylinder are constrained in all directions. Temperature within the cylinder varies as a sinusoidal function of time as given by



Fig. 10.14 Plastic work contours at load step 10 (t=30 s)



Fig. 10.15 Plastic work contours at load step 20 (t=60 s)

$$T(x,y,z,t) = T_{\min} + (T_{\max} - T_{\min}) \sin\left(\frac{\pi t}{t_{fin}}\right) \tag{10.7}$$

in which $T_{\min}$ and $T_{\max}$ are the minimum and maximum values of the applied temperature, t designates time, and t_{fin} is the time at the end of the process. The numerical values of these parameters are given as: $T_{\min} = -35\ ^\circ\mathrm{C}$, $T_{\max} = 125\ ^\circ\mathrm{C}$, and $t_{fin} = 60\ \mathrm{sec}$. The material properties of the solder vary with temperature, as tabulated in Table 10.2. The material exhibits both isotropic hardening plasticity and creep behavior. For the creep behavior, the following strain rate equation is utilized:

$$\dot{\varepsilon}_{cr} = C_1[\sinh(C_2\sigma)]^{C_3} e^{-C_4/T} \tag{10.8}$$

in which $\dot{\varepsilon}_{cr}$ is the creep strain rate, T is the temperature in Kelvin, σ is the equivalent stress, and C_1, C_2, C_3, and C_4 are specific parameters for the generalized Garofalo creep model. In this problem, values of these parameters are taken as $C_1 = 23.3 \times 10^6$ (1/sec), $C_2 = 6.699 \times 10^{-2}$, $C_3 = 3.3$, and $C_4 = 8.12 \times 10^3$ (1/°K). The plastic behavior at different temperatures is shown in Fig. 10.13. The goal is to find strain fields, including plastic and creep strains, at different times.

Model Generation

The problem is solved using two-dimensional 8-noded **PLANE183** elements with the axisymmetric option. In order to specify temperature-dependent elastic properties, a temperature table is constructed using the **MPTEMP** command, followed by the specification of properties using the **MPDATA** command. Note that temperatures are specified in Kelvin. The stress-strain variations for the isotropic hardening model at different temperatures are specified using the **TB**, **TBTEMP**, and **TBPT** commands. In the analysis, millimeter (mm) is used for length dimensions and megapascal (MPa) is used for stresses (also elastic modulus). Thus, the resulting displacements and stresses are in millimeters and megapascals, respectively. Again, this is a special case where an inconsistent unit system works correctly, and it is highly recommended that a unit analysis be performed before applying similar approaches to other mixed-unit systems.

```
/PREP7                        ! ENTER PREPROCESSOR
ET,1,PLANE183                 ! USE PLANE183 ELEMENT
KEYOPT,1,3,1                  ! USE AXISYMMETRY
MPTEMP                        ! INITIALIZE MAT TEMP TABLE
MPTEMP,1,238.15,258.15,278.15       ! CONSTRUCT MAT TEMP
                                    ! TABLE
MPTEMP,4,298.15,323.15,348.15,373.15,398.15
                              ! SPECIFY TEMPERATURE
                              ! DEPENDENT ELASTIC MODULUS
MPDATA,EX,1,1,40781,37825,34884
```

```
MPDATA,EX,1,4,31910,28149,24425,20710,16942
                           !SPECIFY TEMPERATURE
                           !DEPENDENT POISSON'S RATIO
MPDATA,NUXY,1,1,0.354,0.3565,0.36
MPDATA,NUXY,1,4,0.3628,0.365,0.37,0.3774,0.3839
                           !SPECIFY TEMPERATURE
                           !DEPENDENT CTE
MPDATA,ALPX,1,1,2.427E-5,2.448E-5,2.466E-5
MPDATA,ALPX,1,4,2.48E-5,2.501E-5,2.526E-5,
                                        ➥2.552E-5,2.579E-5
MPTEMP                     !INITIALIZE MAT TEMP TABLE
TB,MISO,1,8,7,             !SPECIFY ISOTROPIC HARDENING
TBTEMP,-35+273.15          !SPECIFY TEMPERATURE AS -35 C
TBPT,DEFI,9.313161E-04,37.980    !SPECIFY STRAIN VS
TBPT,DEFI,8.931316E-03,50.677    !STRESS DATA POINTS AT
TBPT,DEFI,1.693132E-02,56.307    !THIS TEMPERATURE
TBPT,DEFI,2.493132E-02,60.142    !
TBPT,DEFI,4.893132E-02,67.620    !
TBPT,DEFI,8.093132E-02,73.889    !
TBPT,DEFI,1.009313E-01,76.836    !
TBTEMP,-15+273.15          !SPECIFY TEMPERATURE AS -15 C
TBPT,DEFI,9.295968E-04,35.162    !SPECIFY STRAIN VS
TBPT,DEFI,8.929597E-03,46.917    !STRESS DATA POINTS AT
TBPT,DEFI,1.692960E-02,52.129    !THIS TEMPERATURE
TBPT,DEFI,2.492960E-02,55.680    !
TBPT,DEFI,4.892960E-02,62.603    !
TBPT,DEFI,8.092960E-02,68.407    !
TBPT,DEFI,1.009296E-01,71.136    !
TBTEMP,5+273.15            !SPECIFY TEMPERATURE AS 5 C
TBPT,DEFI,9.271872E-04,32.344    !SPECIFY STRAIN VS
TBPT,DEFI,8.927187E-03,43.158    !STRESS DATA POINTS AT
TBPT,DEFI,1.692719E-02,47.952    !THIS TEMPERATURE
TBPT,DEFI,2.492719E-02,51.218    !
TBPT,DEFI,4.892719E-02,57.586    !
TBPT,DEFI,8.092719E-02,62.925    !
TBPT,DEFI,1.009272E-01,65.435    !
TBTEMP,25+273.15           !SPECIFY TEMPERATURE AS 25 C
TBPT,DEFI,9.252899E-04,29.526    !SPECIFY STRAIN VS
TBPT,DEFI,8.925290E-03,39.398    !STRESS DATA POINTS AT
TBPT,DEFI,1.692529E-02,43.774    !THIS TEMPERATURE
TBPT,DEFI,2.492529E-02,46.756    !
TBPT,DEFI,4.892529E-02,52.569    !
TBPT,DEFI,8.092529E-02,57.443    !
TBPT,DEFI,1.009253E-01,59.734    !
TBTEMP,50+273.15           !SPECIFY TEMPERATURE AS 50 C
TBPT,DEFI,9.237984E-04,26.004    !SPECIFY STRAIN VS
TBPT,DEFI,8.923798E-03,34.698    !STRESS DATA POINTS AT
TBPT,DEFI,1.692380E-02,38.552    !THIS TEMPERATURE
TBPT,DEFI,2.492380E-02,41.178    !
TBPT,DEFI,4.892380E-02,46.298    !
TBPT,DEFI,8.092380E-02,50.590    !
TBPT,DEFI,1.009238E-01,52.608    !
TBTEMP,75+273.15           !SPECIFY TEMPERATURE AS 75 C
TBPT,DEFI,9.204504E-04,22.482    !SPECIFY STRAIN VS
TBPT,DEFI,8.920450E-03,29.998    !STRESS DATA POINTS AT
TBPT,DEFI,1.692045E-02,33.330    !THIS TEMPERATURE
```

```
TBPT,DEFI,2.492045E-02,35.601     !
TBPT,DEFI,4.892045E-02,40.027     !
TBPT,DEFI,8.092045E-02,43.738     !
TBPT,DEFI,1.009205E-01,45.482     !
TBTEMP,100+273.15              !SPECIFY TEMPERATURE AS 100 C
TBPT,DEFI,9.154515E-04,18.959     !SPECIFY STRAIN VS
TBPT,DEFI,8.915451E-03,25.298     !STRESS DATA POINTS AT
TBPT,DEFI,1.691545E-02,28.108     !THIS TEMPERATURE
TBPT,DEFI,2.491545E-02,30.023     !
TBPT,DEFI,4.891545E-02,33.756     !
TBPT,DEFI,8.091545E-02,36.885     !
TBPT,DEFI,1.009155E-01,38.357     !
TBTEMP,125+273.15              !SPECIFY TEMPERATURE AS 125 C
TBPT,DEFI,9.111675E-04,15.437     !SPECIFY STRAIN VS
TBPT,DEFI,8.911168E-03,20.598     !STRESS DATA POINTS AT
TBPT,DEFI,1.691117E-02,22.886     !THIS TEMPERATURE
TBPT,DEFI,2.491117E-02,24.445     !
TBPT,DEFI,4.891117E-02,27.485     !
TBPT,DEFI,8.091117E-02,30.033     !
TBPT,DEFI,1.009112E-01,31.231     !
C1 = 23.3E6                    !DEFINE CREEP MODEL
C2 = 6.6990E-2                 !PARAMETERS
C3 = 3.300                     !
C4 = 6.7515E4/8.314            !
TB,CREEP,1,,,8                 !SPECIFY GENERALIZED
                               !GAROFALO CREEP MODEL
TBDATA,1,C1,C2,C3,C4           !SPECIFY CREEP MODEL PAR.
RATE,1                         !USE CREEP STRAIN RATE EFFECT
RECTNG,0,10,0,100              !CREATE RECTANGLE
ESIZE,2                        !SPECIFY ELEMENT SIZE
MSHKEY,1                       !ENFORCE MAPPED MESHING
AMESH,ALL                      !MESH AREA
FINISH                         !EXIT PREPROCESSOR
```

Solution

The solution is obtained in twenty load steps. Nonlinear geometry effects are turned on using the **NLGEOM** command. Automatic time stepping is turned on (**AUTOTS** command) so that ANSYS can adjust the time step size values for the substeps within each load step. The loads (temperature change) are interpolated for each substep from the values of the previous load step to the values of the current load step using the **KBC** command. A do loop is utilized for creating load step files, in which the temperature conditions are calculated and applied (**BFUNIF** command) before they are written to load step files (**LSWRITE** command). Finally, the solution is obtained using the **LSSOLVE** command.

```
/SOLU                      !ENTER SOLUTION PROCESSOR
TMIN=238.15                !DEFINE PARAMETER FOR MIN
                           !TEMPERATURE
TMAX=398.15                !DEFINE PARAMETER FOR MAX
                           !TEMPERATURE
```

```
NLS=20                    !DEFINE PARAMETER FOR # OF LOAD
                          !STEPS
TMX=60                    !DEFINE PARAMETER FOR FINAL TIME
PI=4*ATAN(1)              !DEFINE PARAMETER FOR π
ANTYPE,STATIC             !SPECIFY ANALYSIS TYPE AS STATIC
NLGEOM,ON                 !TURN NONLINEAR GEOMETRY EFFECTS
KBC,0                     !ENFORCE RAMPED LOADING
AUTOTS,ON                 !TURN AUTOMATIC TIME STEPPING ON
TREF,273.15               !SPECIFY STRESS-FREE TEMPERATURE
DELTIM,1E-2,1E-3,3        !SPECIFY TIME STEP SIZE PARAMETERS
NSEL,S,LOC,Y,0            !SELECT NODES AT Y = 0
D,ALL,ALL                 !CONSTRAIN ALL DOFS
ALLSEL                    !SELECT EVERYTHING
NSEL,S,LOC,Y,100          !SELECT NODES AT Y = 100
D,ALL,ALL                 !CONSTRAIN ALL DOFS
ALLSEL                    !SELECT EVERYTHING
TIM=TMX/NLS               !FIND END TIME FOR CURRENT LOAD
                          !STEP
*DO,I,1,NLS               !LOOP OVER LOAD STEPS
TIME,TIM                  !SPECIFY TIME FOR CURRENT LOAD
                          !STEP
TC=TMIN+(TMAX-TMIN)*SIN(PI*TIM/TMX)     ! FIND TEMP FOR
                                        ! CURRENT LS
BFUNIF,TEMP,TC            !SPECIFY TEMPERATURE FOR
                          !CURRENT LS
LSWRITE,I                 !WRITE LOAD STEP FILE
TIM=TIM+TMX/NLS           !FIND END TIME FOR NEXT LOAD STEP
*ENDDO                    !END LOOP OVER LOAD STEPS
LSSOLVE,1,NLS             !SOLVE FROM LOAD STEPS
FINISH                    !EXIT SOLUTION PROCESSOR
```

Postprocessing

Once the solution is obtained, the user can review a multitude of results items, including elastic, plastic, creep, and total strain components, as well as the stress components. The command input segment given below obtains contour plots for a few of the items mentioned above, in two different load steps. The results for different load steps are read using the **SET** command. For clarity in the contour plots, the triad and the symbols for minimum and maximum quantities are turned off. Using the **/TRIAD** and **/PLOPTS** commands. The two load steps considered here are the 10th and the last load steps. The 10th load step corresponds to the maximum temperature condition while the last load step brings the structure to its initial configuration. Although initially stress free, after applying a time-dependent temperature load and returning to the initial temperature, the structure experiences residual stresses and strains. Figure 10.16 shows the contour plot of equivalent stress at load step 20, which exhibits a significant amount of stress. Figures 10.17 and 10.18 show contour plots of plastic and creep strains at load step 20, respectively.



Fig. 10.16 Contours of residual equivalent stress at load step 20 (t=60 s)



Fig. 10.17 Contours of plastic strain in the y-direction at load step 20 (t=60 s)



Fig. 10.18 Contours of creep strain in the y-direction at load step 20 (t=60 s)

```
/POST1                 !ENTER GENERAL POSTPROCESSOR
/TRIAD,OFF             !TURN THE TRIAD OFF
/PLOPTS,MINM,0         !TURN MIN & MAX SYMBOLS OFF
SET,10                 !READ RESULTS AT LOAD STEP 10
PLNSOL,NL,PLWK         !PLOT PLASTIC WORK CONTOURS
PLNSOL,S,EQV           !PLOT EQUIVALENT STRESS CONTOURS
PLNSOL,EPPL,Y          !PLOT CONTOURS FOR PLASTIC STRAIN
                       !IN Y-DIR.
PLNSOL,EPCR,Y          !PLOT CONTOURS FOR CREEP STRAIN IN
                       !Y-DIR.
SET,20                 !READ RESULTS AT LOAD STEP 20
PLNSOL,NL,PLWK         !PLOT PLASTIC WORK CONTOURS
PLNSOL,S,EQV           !PLOT EQUIVALENT STRESS CONTOURS
PLNSOL,EPPL,Y          !PLOT CONTOURS FOR PLASTIC STRAIN
                       !IN Y-DIR.
PLNSOL,EPCR,Y          !PLOT CONTOURS FOR CREEP STRAIN IN
                       !Y-DIR.
```

10.3 Contact

Nonlinearity due to contact conditions arises because the prescribed displacements on the boundary depend on the deformation of the structure. Furthermore, no-interpenetration conditions are enforced while the extent of the contact area is unknown.

Contact between two bodies with no bonding (such as glue, solder, or weld) is a challenging problem, mainly stemming from the lack of prior knowledge of the contact regions. Another complication is that, in most of the cases, there is friction between the contacting bodies. Both of these two factors make contact analysis highly nonlinear. In addition to these, if the materials involved exhibit nonlinear material behavior or transient effects, achieving convergence becomes even more difficult.

Before starting a contact analysis, the user needs to be aware of two main considerations: the difference in stiffness of the contacting bodies and the location of possible contact regions. If one of the contacting bodies is significantly stiffer than the other, then the *rigid-to-flexible* contact option can be used in ANSYS, resulting in a considerable reduction in computational time and, possibly, less difficulty in convergence. There are three contact models in ANSYS: node-to-node, node-to-surface, and surface-to-surface. Each of these models requires different types of contact elements. The node-to-node contact model is used when the contact region is accurately known a priori and when the nodes belonging to either contact surfaces are paired, thus requiring these nodes to have same the coordinates. If a large amount of sliding between the contact surfaces is expected, node-to-node contact is not suitable. The node-to-surface contact model is used when a specific point on one of the surfaces, e.g., a corner, is expected to make contact with a rather smooth surface. This model does not require accurate a priori knowledge of the contact region, and the mesh pattern on either surface does not need to be compatible. The surface-to-surface contact model is used when contact regions are not known accurately and a significant amount of sliding is expected. In this model, one of the surfaces is called the *contact surface* and the other, the *target surface*.

In a typical ANSYS contact analysis, the model is first meshed with conventional elements (beam, plane, or solid), and then the contact elements are created along the potential contact regions. Different element types are used for different contact models.

A contact analysis can be highly complicated, requiring the user to have a good understanding beforehand. It is recommended that the user read the section entitled “Contact” in the ANSYS Structural Analysis Guide as it provides a highly detailed description on how to perform a contact analysis in ANSYS, as well as several helpful hints to achieve converged solutions.

In the following subsections, two contact problems are used to demonstrate a contact analysis within ANSYS. In the first problem, the node-to-surface contact model is used in the simulation of a block dropping on a beam. Thus, the contact analysis is combined with dynamics, leading to time-dependent results.



Fig. 10.19 Schematic of the block and beam (*left*), and the corresponding finite element mesh (*right*)

The second problem is the simulation of a nano-indentation test of a thin film deposited on a hard substrate. The problem is solved using surface-to-surface contact elements, and the thin film exhibits an elastic-perfectly plastic material behavior.

10.3.1 Contact Analysis of a Block Dropping on a Beam

Consider a 6-in × 6-in × 1-in block free falling onto a 100-in-long beam, as shown in Fig. 10.19. The block and the beam are made of the same material, with elastic modulus $E = 1\times10^6$ psi and density $\rho = 0.001$ lb/in^3. The beam has a cross-sectional area of 0.5 in^2, an area moment of inertia of $I_{yy} = 0.05$ in^4, and a height of 1 in. The block is initially at 100 in above the beam, with its center point 9 in to the right of the mid-point of the beam. The goal is to obtain the time-dependent response of the beam and the block.

Model Generation

The problem is solved using two-dimensional elements for both the beam (**`BEAM188`**) and the block (**`PLANE182`**). In modeling the block, plane stress with thickness idealization is used. Between the block (contact surface) and the beam (target surface), contact elements are defined using element types **`CONTAC172`** and **`TARGE169`**. This is achieved by first defining two *components*, one containing the nodes of the target surface and the other containing the nodes of the contact surface.

```
/PREP7                    ! ENTER PREPROCESSOR
ET,1,BEAM188              ! ELEMENT TYPE 1 IS BEAM188
KEYOPT,1,3,3              ! SPECIFY CUBIC FORM
ET,2,PLANE182,,,3         ! ELEMENT TYPE 2 IS PLANE182
ET,3,TARGE169             ! ELEMENT TYPE 1 IS TARGE169
ET,4,CONTA172             ! ELEMENT TYPE 2 IS CONTA172
KEYOPT,4,10,2             ! UPDATE CONTACT STIFFNESS AT EACH
                          ! ITERATION
R,1,1                     ! REAL CONSTANT SET 2 FOR PLANE182
SECTYPE,1,BEAM,RECT       ! DEFINE A RECTANGULAR CROSS-
                          ! SECTION FOR BEAM
SECDATA,0.5,1             ! DEFINE SECTION GEOMETRY DATA
MP,EX,1,1E6               ! SPECIFY ELASTIC MODULUS
MP,DENS,1,.001            ! SPECIFY DENSITY
K,1,0,0                   ! CREATE KEYPOINT 1 (LEFT END OF
                          ! BEAM)
K,2,100,0                 ! CREATE KEYPOINT 2 (RIGHT END OF
                          ! BEAM)
L,1,2                     ! CREATE LINE FOR BEAM
ESIZE,,10                 ! USE 10 ELEMENTS PER LINE
LMESH,1                   ! MESH THE LINE WITH BEAM3 ELEMENTS
RECTNG,56,62,100,106      ! CREATE RECTANGULAR AREA
ESIZE,,1                  ! USE 1 ELEMENT PER LINE
TYPE,2                    ! SWITCH TO ELEMENT TYPE 2
                          ! (PLANE182)
REAL,1                    ! SWITCH TO REAL CONSTANT SET 1
AMESH,ALL                 ! MESH THE RECTANGULAR AREA
R,2                       ! REAL CONSTANT SET 2 FOR CONTACT
                          ! PAIRS
REAL,2                    ! SWITCH TO REAL CONSTANT SET 2
                          ! GENERATE THE TARGET SURFACE
LSEL,S,,,1                ! SELECT LINE 1
CM,TARGET,LINE            ! DEFINE COMPONENT NAMED "TARGET"
TYPE,3                    ! SWITCH TO ELEMENT TYPE 3
                          ! (TARGE169)
NSLL,S,1                  ! SELECT NODES ASSOCIATED WITH THE
                          ! SELECTED LINE
ESLN,S,0                  ! SELECT THE ELEMENTS ATTACHED TO
                          ! THOSE NODES
ESURF                     ! GENERATE ELEMENTS OVERLAID ON
                          ! THE FREE FACES OF EXISTING
                          ! SELECTED ELEMENTS
                          ! GENERATE THE CONTACT SURFACE
LSEL,S,,,2,5              ! SELECT LINES 2,3,4, AND 5
CM,CONTACT,LINE           ! DEFINE COMPONENT NAMED "TARGET"
TYPE,4                    ! SWITCH TO ELEMENT TYPE 4
                          ! (CONTA172)
NSLL,S,1                  ! SELECT NODES ASSOCIATED WITH THE
                          ! SELECTED LINES
ESLN,S,0                  ! SELECT THE ELEMENTS ATTACHED TO
                          ! THOSE NODES
ESURF                     ! GENERATE ELEMENTS OVERLAID ON
                          ! THE FREE FACES OF EXISTING
                          ! SELECTED ELEMENTS
ALLSEL                    ! SELECT EVERYTHING
```

Solution

The transient solution is obtained in two load steps. The first load step encompasses a very short duration (0.002 s) and is solved using two substeps without time integration. This is done in order to set up the initial conditions for the transient solution. In the second load step, time integration is turned on (**TIMINT** command) and so is the automatic time stepping option (**AUTOTS** command). Also, the predictor option is turned on before the solution is obtained for the second load step using the **PRED** command. This option allows ANSYS to make a prediction of the displacements at the beginning of each substep, thus improving convergence. Structural damping is specified in terms of viscous damping, which produces a damping matrix in the form $[\mathbf{C}] = \alpha\,[\mathbf{M}] + \beta\,[\mathbf{K}]$, in which $[\mathbf{C}]$, $[\mathbf{M}]$, and $[\mathbf{K}]$ are the damping, mass, and stiffness matrices, respectively. In this case, only the damping through stiffness matrix is enforced by specifying the stiffness matrix multiplier β using the **BETAD** command.

```
/SOLU                    ! ENTER SOLUTION PROCESSOR
ANTYPE,TRANS             ! DECLARE ANALYSIS TYPE A
                         ! TRANSIENT
NLGEOM,ON                ! TURN NONLINEAR GEOMETRY EFFECTS
                         ! ON
LUMPM,ON                 ! USE LUMPED MASS ASSUMPTION
KSEL,S,KP,,1,2           ! SELECT KEYPOINTS 1 AND 2
NSLK,S                   ! SELECT NODES ATTACHED TO
                         ! KEYPOINTS
D,ALL,ALL,0              ! CONSTRAIN ALL DOFS AT SELECTED
                         ! NODES
ALLSEL,ALL               ! SELECT EVERYTHING
ESEL,S,ENAME,,182        ! SELECT ELEMENTS OF TYPE PLANE182
NSLE,S                   ! SELECT NODES ATTACHED TO SELECTED
                         ! ELEMENTS
D,ALL,ALL,0              ! CONSTRAIN ALL DOFS AT SELECTED
                         ! NODES
ALLSEL,ALL               ! SELECT EVERYTHING
ACEL,,386                ! DEFINE GRAVITATIONAL ACCELERATION
TIME,.0002               ! SPECIFY TIME AT THE END OF 1ST
                         ! LOAD STEP
DELTIM,.0001             ! SPECIFY TIME STEP SIZE
KBC,1                    ! ENFORCE STEPPED LOADING
BETAD,.000318            ! STIFFNESS MATRIX MULTIPLIER FOR
                         ! DAMPING
TIMINT,OFF               ! TURN OFF TIME INTEGRATION
CNVTOL,F,,.00001         ! SPECIFY FORCE CONVERGENCE
                         ! TOLERANCE
OUTRES,ALL,LAST          ! SAVE ONLY THE LAST RESULTS SET
SOLVE                    ! OBTAIN SOLUTION FOR LOAD STEP 1
ESEL,S,ENAME,,182        ! SELECT ELEMENTS OF TYPE PLANE182
NSLE,S                   ! SELECT NODES ATTACHED TO SELECTED
                         ! ELEMENTS
DDELE,ALL,ALL            ! DELETE DOF CONSTRAINTS ON
```

```
                                ! SELECTED NODES
ALLSEL,ALL                      ! SELECT EVERYTHING
TIME,3                          ! TIME AT THE END OF 2ND LS AS
                                ! 3 SEC
DELTIM,.02,.0002,.02            ! USE INITIAL TIME STEP SIZE 0.02
                                ! WITH MINIMUM 0.0002 AND MAXIMUM
                                ! 0.02
AUTOTS,ON                       ! TURN ON AUTOMATIC TIME STEPPING
TIMINT,ON                       ! TURN ON TIME INTEGRATION
CNVTOL,F                        ! RESET FORCE CONVERGENCE TOLERANCE
PRED,ON                         ! ACTIVATE PREDICTOR FOR INITIAL
                                ! GUESS
OUTRES,ALL,ALL                  ! SAVE RESULTS FOR EVERY SUBSTEP
SOLVE                           ! OBTAIN SOLUTION
FINISH                          ! EXIT SOLUTION PROCESSOR
```

Postprocessing

After the solution is obtained, results are reviewed in both the *General Postprocessor* (results associated with the whole structure at a specific time) and *Time History Postprocessor* (results associated with a specific node in the structure along the entire time domain). First, in the *Time History Postprocessor*, the displacements in the *y*-direction at the mid-point of the beam (parameter **MID**) and at four corners of the block (parameters **BL**, **BR**, **TR**, and **TL**) are stored (**NSOL** command) in user-defined parameters, and they are plotted against time (**PLVAR** command), as shown in Fig. 10.20. Similarly, the reaction forces in the *y*-direction at both ends of the beam are stored (**RFOR** command) in user-defined parameters (parameters **FL** and **FR**), and they are plotted against time (Fig. 10.21). In both of these graphs,



Fig. 10.20 Time variation of *y*-displacements at four corners of the block and at the midpoint of the beam



Fig. 10.21 Time variation of reaction forces in the beam in the *y*-direction



Fig. 10.22 Deformed configuration at time 1.742 s



Fig. 10.23 Moment diagram of the beam at time 1.742 s

labels and ranges for the *x*- and *y*-axes are modified using the **/AXLAB**, **/AUTO**, **/XRANGE**, and **/YRANGE** commands. After reviewing the responses of specific nodes as functions of time, the deformed shape and moment diagram at a specific time (1.742 s) are reviewed in the *General Postprocessor*. The results associated with time 1.742 s are read using the **SET** command, and the deformed shape is obtained using the **PLDISP** command (Fig. 10.22). The moment results associated with nodes **I** and **J** of each element are stored in element table items **MOMZI** and **MOMZJ**, respectively, using the **ETABLE** command. Finally, the **PLLS** command is used for plotting the moment diagram (Fig. 10.23). The command input block below includes a **/WAIT** command after each plot command (**PLVAR** and **PLDISP** commands), which causes ANSYS to suspend operations for a specified duration (2 s in this case).

```
/POST26                       !ENTER TIME HISTORY
                              !POSTPROCESSOR
NSOL,2,6,U,Y,MID              !STORE MID-PT BEAM Y-DISP TO
                              !MID
NSOL,3,12,U,Y,BL              !STORE BOTTOM LEFT BLOCK Y-DISP
                              !TO BL
NSOL,4,13,U,Y,BR              !STORE BOTTOM RIGHT BLOCK Y-
                              !DISP TO BR
NSOL,5,14,U,Y,TR              !STORE TOP RIGHT BLOCK Y-DISP
                              !TO TR
NSOL,6,15,U,Y,TL              !STORE TOP LEFT BLOCK Y-DISP TO
                              !TL
/AXLAB,Y,DISPLACEMENT         !SPECIFY Y AXIS LABEL FOR GRAPH
/AXLAB,X,TIME                 !SPECIFY X AXIS LABEL FOR GRAPH
/AUTO,1                       !AUTOMATIC FOCUS/DISTANCE FOR
                              !GRAPH
PLVAR,2,3,4,5,6               !PLOT VARIABLES 2-6 VS TIME
/WAIT,2                       !WAIT 2 SECONDS BEFORE
                              !PROCEEDING
RFOR,7,1,F,Y,FL               !STORE LEFT Y-REACTION FORCE TO
                              !FL
RFOR,8,2,F,Y,FR               !STORE RIGHT Y-REACTION FORCE
                              !TO FR
/YRANGE                       !RESET Y AXIS RANGE
/AXLAB,Y,REACTION FORCE       !SPECIFY Y AXIS LABEL FOR GRAPH
PLVAR,7,8                     !PLOT VARIABLES 7 & 8 VS TIME
/WAIT,2                       !WAIT 2 SECONDS BEFORE
                              !PROCEEDING
FINISH                        !EXIT TIME HISTORY
                              !POSTPROCESSOR
```

```
/POST1                        !ENTER GENERAL POSTPROCESSOR
SET,,,,,1.742                 !SET RESULTS FOR TIME 1.742 SEC
ETABLE,MOMZI,SMISC,3          !ELEM TABLE FOR MOMENT ON NODE
                              !I
ETABLE,MOMZJ,SMISC,16         !ELEM TABLE FOR MOMENT ON NODE
                              !J
/DSCALE,1,1                   !DO NOT SCALE DISPLACEMENTS
/TRIAD,OFF                    !TURN OFF THE TRIAD
PLDISP,1                      !PLOT DEFORMED SHAPE
/WAIT,2                       !WAIT 2 SECONDS BEFORE
                              !PROCEEDING
ESEL,S,TYPE,,1                !SELECT BEAM ELEMENTS (TYPE 1)
NSLE,S                        !SELECT NODES ATTACHED TO
                              !ELEMENTS
PLLS,MOMZI,MOMZJ              !PLOT ELEM TABLE ITEMS ON MESH
ALLSEL,ALL                    !SELECT EVERYTHING
FINISH                        !EXIT GENERAL POSTPROCESSOR
```

In order to obtain an animation of the time-dependent displacement response of the system:

- Read the results associated with load step 2 (**`SET`** command) using the following menu path:

Main Menu > General Postproc > Read Results > Last Set

- Obtain the deformed shape (**`PLDISP`** command) using the following menu path:

Main Menu > General Postproc > Plot Results > Deformed Shape

- *Plot Deformed Shape* dialog box appears; select ***Def + undef edge*** radio-button; click on ***OK***.
- Create animation using the following menu path:

Utility Menu > PlotCtrls > Animate > Over Time

- *Animate Over Time* dialog box appears; enter ***100*** for ***Number of animation frames***. Click on ***OK*** and wait until *Animation Controls Window* appears.

10.3.2 Simulation of a Nano-Indentation Test

Nano-indentation tests are commonly used for evaluation of the response of thin films. Consider a sol-gel layer deposited on a glass substrate, which is indented by means of a conical diamond indenter as shown in Fig. 10.24. The thickness values for the sol-gel film and the glass substrate are $2\,\mu\mathrm{m}$ and $6\,\mu\mathrm{m}$, respectively. The indenter has an angle of 68° measured from the axis of rotation. In order to correctly simulate the indentation phenomenon, a contact analysis is utilized. For this purpose, target elements (**`TARGE170`**) are placed along the top surface of the film and contact elements (**`CONTA174`**) are used along the bottom surface of the indenter. The indentation is simulated by applying displacement boundary conditions in the *y*-direction to the nodes along the bottom surface of the substrate. Consequently, the top surface of the film is pressed against the bottom surface of the indenter, thus exerting the contact elements against the target elements. The contact is assumed to be frictionless. The indentation is performed using several displacement steps, each of which is written to a load step file. Both loading and unloading are simulated. Displacement step sizes for loading and unloading are $0.04\,\mu\mathrm{m}$ and $0.03\,\mu\mathrm{m}$, respectively. All three materials are modeled using **`SOLID185`** elements. Since the problem possesses symmetry with respect to the *y*-axis, only one octant ($1/8^{\mathrm{th}}$) of the geometry is modeled. Normal displacements are constrained along the symmetry planes, and *x*- and *z*-displacements are constrained along the axis of rotation. The top surface of the indenter is constrained in all directions. The sol-gel film exhibits elastic-perfectly plastic behavior with yield stress of $700\ \mathrm{MPa}$ while the diamond indenter and the glass substrate are both elastic. The elastic modulus and Poisson's ratio values for the constituent materials are given in Table 10.4. The goal is to obtain indentation vs. force response for the film.



Fig. 10.24 Sol-gel film on glass substrate, and diamond indenter

Model Generation

In the solid modeling phase, a *bottom-up approach* is used. The solid model is created, starting with keypoints, then lines, areas, and volumes. The elastic-perfectly plastic behavior of the thin film is incorporated using the bilinear isotropic hardening rule (**`TB`** command with **`BISO`** option) with a tangent modulus of zero. The model is first generated in the *x*-*y* plane and meshed using two-dimensional elements (**`PLANE182`**), after which the **`VROTAT`** command is used to generate

Table 10.4 Material properties used in nano-indentation simulation

	Elastic modulus (GPa)	Poisson’s ratio	Material reference number
Glass substrate	75	0.23	1
Sol-gel film	4.5	0.35	2
Diamond indenter	1141	0.07	3



Fig. 10.25 Isometric view of the finite element mesh used for the nano-indentation simulation

the three-dimensional mesh by rotating the meshed areas about the *y*-axis by 45°. Before issuing the **VROTAT** command, the default element type attribute must be changed to the one for three-dimensional elements (element type ***2*** for **SOLID185** in this case). Figure 10.25 shows an isometric view of the mesh used in this analysis.

```
/FILNAM,NANO            ! SPECIFY JOBNAME
/PREP7                  ! ENTER PREPROCESSOR
ET,1,182                ! ELEMENT TYPE 1 IS PLANE182
ET,2,185                ! ELEMENT TYPE 2 IS SOLID185
ET,3,TARGE170           ! ELEMENT TYPE 3 IS TARGE170
ET,4,CONTA174           ! ELEMENT TYPE 4 IS CONTA174
MP,EX,1,75000           ! GLASS SUBSTRATE MAT PROPS
MP,NUXY,1,0.23          !
MP,EX,2,4500            ! SOL GEL FILM MAT PROPS
MP,NUXY,2,0.35          !
TB,BISO,2,1             ! BILINEAR ISOTROPIC HARDENING RULE
TBTEMP,0                !
TBDATA,1,700,0          ! YIELD STRENGTH
MP,EX,3,1141000         ! DIAMOND INDENTER MAT PROPS
MP,NUXY,3,0.07          !
K,1                     ! CREATE KEYPOINTS
K,2,4
K,3,10
K,4,,6
K,5,4,6
K,6,10,6
K,7,,7.2
K,8,4,7.2
K,9,10,7.2
K,10,,8
K,11,4,8
```

```
K,12,10,8
L,1,2                   !CREATE LINES
L,2,3
L,4,5
L,5,6
L,7,8
L,8,9
L,10,11
L,11,12
L,1,4
L,2,5
L,3,6
L,4,7
L,5,8
L,6,9
L,7,10
L,8,11
L,9,12
AL,1,3,9,10             !CREATE AREAS
AL,2,4,10,11
AL,3,5,12,13
AL,4,6,13,14
AL,5,7,15,16
AL,6,8,16,17
LESIZE,7,,,15           !SPECIFY LINE DIVISIONS
LESIZE,5,,,15
LESIZE,3,,,15
LESIZE,1,,,15
LESIZE,8,,,10,4
LESIZE,6,,,10,4
LESIZE,4,,,10,4
LESIZE,2,,,10,4
LESIZE,9,,,10,1/4
```

```
LESIZE,10,,,10,1/4
LESIZE,11,,,10,1/4
LESIZE,12,,,8
LESIZE,13,,,8
LESIZE,14,,,8
LESIZE,15,,,8
LESIZE,16,,,8
LESIZE,17,,,8
MSHKEY,1                !ENFORCE MAPPED MESHING
TYPE,1                  !SWITCH TO ET 1
MAT,1                   !SWITCH TO MATERIAL 1
AMESH,1,6               !CREATE MESH
TYPE,2                  !SWITCH TO ET 2
ESIZE,,4                !SPECIFY # OF ELEMS FOR SWEEP
VROTAT,1,2,3,4,5,6,1,10,45   ! SWEEP AREAS TO CREATE
                             ! VOLUME
NSEL,S,LOC,Y,6,8        !SELECT NODES AT 6 μm ≤ y ≤ 8 μm
ESLN,S,1                !SELECT ELEMENTS WITH SELECTED
                        !NODES
EMODIF,ALL,MAT,2        !MODIFY ELEMS TO BE GLASS
TYPE,4                  !SWITCH TO ET 4
AMESH,22                !CREATE CONTACT AREA MESH
ESEL,S,TYPE,,4          !SELECT ELEMENTS WITH ET 4
ESURF,,REVE             !ADJUST OUTWARD NORMAL OF ELEMENTS
                        !GENERATE SOLID MODEL FOR INDENTER
K,23,,8                 !CREATE KEYPOINTS
K,26,,10.5
K,24,3.7,9.5
K,25,3.7,10.5
L,23,24                 !CREATE LINES
L,24,25
L,25,26
L,26,23
AL,40,41,42,43          !CREATE AREA
LESIZE,41,,,5           !SPECIFY LINE DIVISIONS
LESIZE,43,,,5
LESIZE,40,,,15
LESIZE,42,,,15
TYPE,1                  !SWITCH TO ET 1
MAT,3                   !SWITCH TO MATERIAL 3
AMESH,27                !CREATE MESH FOR INDENTER
TYPE,2                  !SWITCH TO ET 2
ESIZE,,4                !SPECIFY # OF ELEMS FOR SWEEP
VROTAT,27,,,,,,23,26,45 !SWEEP AREAS TO CREATE VOLUME
R,1                     !REAL CONSTANT SET 1 FOR CONTACT
                        !PAIR
REAL,1                  !SWITCH TO REAL CONSTANT SET 1
                        !GENERATE THE TARGET SURFACE
ASEL,S,,,22             !SELECT AREA 22
CM,TARGET,AREA          !DEFINE COMPONENT NAMED "TARGET"
TYPE,3                  !SWITCH TO ET 3
NSLA,S,1                !SELECT NODES ASSOCIATED WITH THE
                        !SELECTED AREA
ESLN,S,0                !SELECT THE ELEMENTS ATTACHED TO
                        !THOSE NODES
ESURF                   !GENERATE ELEMENTS OVERLAID ON
                        !THE FREE FACES OF EXISTING
```

```
                               ! SELECTED ELEMENTS
                               ! GENERATE THE CONTACT SURFACE
ASEL,S,,,28                    ! SELECT LINES 2,3,4, AND 5
CM,CONTACT,AREA                ! DEFINE COMPONENT NAMED "TARGET"
TYPE,4                         ! SWITCH TO ET 4
NSLA,S,1                       ! SELECT NODES ASSOCIATED WITH THE
ESLN,S,0                       ! SELECT THE ELEMENTS ATTACHED TO
                               ! THOSE NODES
ESURF                          ! GENERATE ELEMENTS OVERLAID ON
                               ! THE FREE FACES OF EXISTING
                               ! SELECTED ELEMENTS
ALLSEL                         ! SELECT EVERYTHING
FINISH                         ! EXIT PREPROCESSOR
```

Solution

As mentioned previously, the top surface of the diamond indenter is constrained in all directions. In addition, the vertical planes are not allowed to move in their normal direction due to the symmetry boundary conditions imposed by using the **DSYM** command. The indentation is simulated by prescribing a uniform *y*-displacement along the bottom surface of the glass substrate, which causes the film to contact the indenter, starting the indentation process. This displacement is applied in increments of $0.04\,\mu\text{m}$ for the loading phase, and its values are stored in the array parameter **DIS**. The array parameter is created/initialized using the ***DIM** command, with an array size of 57. The first 37 entries of this array correspond to the loading phase displacements and the following 20 entries are for the unloading phase displacements. Once all the displacement values are stored in the array **DIS**, 57 load step files are written (**LSWRITE** command) utilizing a do loop. Nonlinear geometry effects are turned on (**NLGEOM** command), as well as automatic time stepping (**AUTOTS** command). A full Newton-Raphson method is utilized without the *adaptive descent* option (**NROPT** command). The maximum number of equilibrium iterations in each substep is set to 100 using the **NEQIT** command. Prior to initiating the solution using the **LSSOLVE** command, the two-dimensional elements must be unselected, as the sole purpose of their existence is for meshing.

```
/SOLU                          ! ENTER SOLUTION PROCESSOR
SOLCONTROL,0                   ! TURN OFF SOLUTION CONTROLS
ANTYPE,STATIC                  ! SPECIFY ANALYSIS TYPE AS STATIC
NSEL,S,LOC,Z                   ! SELECT NODES AT Z = 0
DSYM,SYMM,Z,0                  ! APPLY SYMMETRY BC ABOUT X-Y PLANE
CLOCAL,11,,,,,,,45             ! DEFINE LOCAL CS
NSEL,S,LOC,Z                   ! SELECT NODES AT Z = 0 (IN LOCAL
                               ! CS)
DSYM,SYMM,Z,11                 ! APPLY SYMMETRY BC IN LOCAL CS
CSYS,0                         ! SWITCH TO GLOBAL CARTESIAN CS
NSEL,S,LOC,Y,10.5              ! SELECT NODES AT Y = 10.5
D,ALL,UY,0                     ! CONSTRAIN Y-DISP AT SELECTED
```

```
                              ! NODES
NSEL,S,LOC,X                  ! SELECT NODES AT X = 0
NSEL,R,LOC,Z                  ! RESELECT NODES AT Z = 0
D,ALL,UX                      ! CONSTRAIN X-DISP AT SELECTED
                              ! NODES
D,ALL,UZ                      ! CONSTRAIN Z-DISP AT SELECTED
                              ! NODES
*DIM,DIS,ARRAY,57             ! INITIALIZE ARRAY PARAMETER DIS
A=0                           ! INITIALIZE PARAMETER A
*DO,I,1,36                    ! START DO LOOP FOR LOADING
DIS(I)=A                      ! STORE DISP VALUE IN DIS FOR
                              ! CURRENT LS
A=A+0.04                      ! UPDATE PARAMETER A
*ENDDO                        ! END DO LOOP ON LOADING
B=0.03                        ! INITIALIZE PARAMETER B
*DO,I,37,57                   ! START DO LOOP FOR UNLOADING
DIS(I)=A-B                    ! STORE DISP VALUE IN DIS FOR
                              ! CURRENT LS
B=B+0.03                      ! UPDATE PARAMETER B
*ENDDO                        ! END DO LOOP ON UNLOADING
NLGEOM,ON                     ! TURN ON NONLINEAR GEOMETRY
                              ! EFFECTS
AUTOTS,ON                     ! TURN ON AUTOMATIC TIME STEPPING
OUTRES,,1                     ! SAVE RESULTS FOR LAST SUBSTEP OF
                              ! EACH LS
NROPT,FULL,,OFF               ! USE FULL NEWTON-RAPHSON WITH NO
                              ! ADAPTIVE DESCENT
NEQIT,100                     ! USE MAXIMUM 100 EQUILIBRIUM
                              ! ITERATIONS
ALLSEL                        ! SELECT EVERYTHING
*DO,I,1,57                    ! START DO LOOP FOR WRITING LOAD
                              ! STEPS
NSEL,S,LOC,Y                  ! SELECT NODES AT Y = 0
D,ALL,UX                      ! CONSTRAIN X-DISP AT SELECTED
                              ! NODES
D,ALL,UZ                      ! CONSTRAIN Z-DISP AT SELECTED
                              ! NODES
D,ALL,UY,DIS(I)               ! SPECIFY Y-DISP ALONG THE BOTTOM
                              ! SURFACE OF THE SUBSTRATE
ALLSEL                        ! SELECT EVERYTHING
LSWRITE,I                     ! WRITE LOAD STEP FILE
*ENDDO                        ! END DO LOOP ON WRITING LOAD STEPS
ESEL,U,TYPE,,1                ! UNSELECT ELEMENT TYPE 1
LSSOLVE,1,57                  ! SOLVE FROM LS FILES (1 TO 57)
FINISH                        ! EXIT SOLUTION PROCESSOR
```

Postprocessing

With the goal of obtaining the load vs. indentation depth response of the sol-gel thin film deposited on a glass substrate, the following command input segment is used. Load and indentation depth values for each load step are extracted within a do loop,



Fig. 10.26 Load vs. indentation depth for sol-gel film on glass substrate

in which the results associated with the current load step are read using the **SET** command. The parameter **SUM** is the total reaction force in the *y*-direction along the bottom surface of the glass substrate. This quantity must be identical to the force exerted by the indenter to the top surface of the film because of equilibrium of forces. The command ***GET** is used on numerous occasions to retrieve model and results information about the nodes. For each load step, the applied indentation depth from array parameter **DIS**, and the corresponding total reaction force stored in parameter **SUM** are written to the file **NANO_RF_D.OUT** using a combination of **/OUTPUT *VWRITE** and commands. Figure 10.26 shows the resulting load vs. indentation depth response.

```
/POST1                          ! ENTER GENERAL POSTPROCESSOR
*DO,J,1,57                      ! LOOP OVER LOAD STEPS
SET,J                           ! READ RESULTS SET
NSEL,S,LOC,Y                    ! SELECT NODES AT Y = 0
*GET,NUMNOD,NODE,0,COUNT        ! STORE # OF NODES IN NUMNOD
*GET,CURNOD,NODE,0,NUM,MIN      ! STORE MIN NODE # TO CURNOD
SUM=0                           ! INITIALIZE TOTAL REACTION F
*DO,I,1,NUMNOD                  ! LOOP OVER SELECTED NODES
*GET,RFY,NODE,CURNOD,RF,FY      ! STORE Y REACT. FORCE IN RFY
SUM=SUM+RFY                     ! UPDATE TOTAL REACTION FORCE
CURNOD=NDNEXT(CURNOD)           ! UPDATE CURRENT NODE NUMBER
*ENDDO                          ! END LOOP OVER SELECTED NODES
DISJ=DIS(J)                     ! STORE CURRENT INDENTATION
                                ! DEPTH TO PARAMETER DISJ
SUM=SUM/100                     ! CONVERT TO MILLINEWTONS
/OUTPUT,NANO_RF_D,OUT,,APPEND ! REDIRECT OUTPUT TO FILE
*VWRITE,DISJ,SUM                ! WRITE DISJ AND SUM TO FILE
(E16.8,5X,E16.8)                ! FORMAT STATEMENT
/OUTPUT                         ! REDIRECT OUTPUT TO OUTPUT W
*ENDDO                          ! END LOOP OVER LOAD STEPS
```